

# SM\_tutorial-J-Rev,2023

June 9, 2023

## 1 Sage 多様体演習 Sage manifold tutorial

このワークシートは SageMath における微分可能多様体の取り扱いについての簡単な紹介です。以下で紹介するツールは [SageManifolds](#) プロジェクト (version 1.2, included in SageMath 8.2). を通じて実装されました。ipynb 形式のワークシートファイルは[ここ](#)からダウンロード可能です。このワークシートを実行するには、コマンド `sage -n jupyter` などを使って SageMath の Jupyter ノートブックインタフェースを起動する必要があります。

[訳注: この日本語版翻訳は山本昇によって作成されました。翻訳者へのご連絡は、<https://souran.kek.jp/kss/top/> を経由して、職員情報詳細ページ内の「メール送信」リンクをご利用ください。この翻訳版作成にあたり、東海大学理学部北林さんには貴重なコメントをいただきました。この場を借りてお礼申し上げます。(2023/05/05 追記: 原文の更新を反映しました。)]

This notebook provides a short introduction to differentiable manifolds in SageMath. The tools described below have been implemented through the [SageManifolds](#) project.

Click [here](#) to download the worksheet file (ipynb format). To run it, you must start SageMath with the Jupyter notebook, via the command `sage -n jupyter`

[Note for Japanese version.: this document was translated into Japanese by Noboru Yamamoto. You can reach him through <https://souran.kek.jp/kss/top/>. “Staff Detail Information” page includes a link “send email”. The translator thanks Prof. Kitabayashi of Tokai University for his valuable comments on the draft of the documents. (last update: 2023/05/05)]

このワークシートは SageMath のバージョン 9.2 (あるいはそれ以降) を必要としています、それ以前のバージョンでは SageManifolds の全ての機能が利用可能というわけではないからです。[訳注: この jupyter notebook は SageMath 9.8 を使って実行しています。]

This notebook is valid for version 9.2 or higher of SageMath:

```
[1]: version()
```

```
[1]: 'SageMath version 10.0, Release Date: 2023-05-20'
```

最初に、ノートブックが結果を LaTeX を使った数式表示で表示するようにします。

First we set up the notebook to display mathematical objects using LaTeX rendering:

```
[2]: %display latex
```

## 1.1 多様体の定義 Defining a manifold

例として、 $\mathbb{R}$  上の 3 次元の微分可能多様体を定義します。

As an example let us define a differentiable manifold of dimension 3 over  $\mathbb{R}$ :

```
[3]: M = Manifold(3, 'M', latex_name=r'\mathcal{M}', start_index=1)
```

- 最初の引数、3, は多様体の次元です。SageManifolds では任意の正の整数を指定できます。
- 2 番目の引数、'M', は多様体の名前を与える文字列です。この名前は = 記号の左辺にあるシンボルの名前 (ここでは M) と異なっていても構いません。左辺のシンボル名はコンピュータメモリのなかの多様体オブジェクトを参照しています。一方、引数であたえる名前は、多様体に与えられた数学的な名前です。
- 省略可能な引数、`latex_name=r'\mathcal{M}'`, はこの多様体を LaTeX で表示する際に使う名前を指定します。最初の引用符の前の 'r' に注意してください。: これはこの文字列が raw 文字列であることを示しています。それによって文字列 `\mathcal` 中のバックスラッシュ '\ ' は通常の文字として取り扱われます ('\\' としてエスケープする必要がないということです)。この引数 (`latex_name`) が省略された場合は、第 2 引数に与えられた名前 (ここでは M です) が使われます。
- 次の省略可能な引数 `start_index=1` は多様体のテンソル成分の指標の範囲を指定しています。これを 1 とすることは指標の範囲が  $\{1, 2, 3\}$  であることを示しています。この引数の既定値は `start_index=0` です。
- The first argument, 3, is the manifold dimension. In SageManifolds, it can be any positive integer.
- The second argument, 'M', is a string defining the manifold's name; it may be different from the symbol set on the left-hand side of the = sign (here M): the latter stands for a mere Python variable, which refers to the manifold object in the computer memory, while the string 'M' is the mathematical symbol chosen for the manifold.
- The optional argument `latex_name=r'\mathcal{M}'` sets the LaTeX symbol to display the manifold. Note the letter 'r' in front on the first quote: it indicates that the string is a raw one, so that the backslash character in `\mathcal` is considered as an ordinary character (otherwise, the backslash is used to escape some special characters). If the argument `latex_name` is not provided by the user, it is set to the string used as the second argument (here 'M')
- The optional argument `start_index=1` defines the range of indices to be used for tensor components on the manifold: setting it to 1 means that indices will range in  $\{1, 2, 3\}$ . The default value is `start_index=0`.

規定のベースフィールド (基底) は  $\mathbb{R}$  であることに注意してください。もし、`field='complex'`, が与えられていれば、複素数体  $\mathbb{C}$  を基底として持つ多様体が定義されます。[[list of all options](#) を参照のこと]

Note that the default base field is  $\mathbb{R}$ . If we would have used the optional argument `field='complex'`, we would have defined a manifold over  $\mathbb{C}$ . See the [list of all options](#) for more details.

M を要求すると、その LaTeX のシンボルとして表示されます。

If we ask for M, it is displayed via its LaTeX symbol:

```
[4]: M
```

[4]:  $\mathcal{M}$

代わりに `print` 関数を使うと、簡潔な記述が表示されます。

If we use the `print` function instead, we get a short description of the object:

```
[5]: print(M)
```

```
3-dimensional differentiable manifold M
```

`show()` 関数を使って、LaTeX を使って表示することを明示することもできます。

```
[6]: show(M)
```

$\mathcal{M}$

`type` コマンドを使うと、 $\mathcal{M}$  に対応する Python オブジェクトのタイプ (`DifferentiableManifold_with_category`) が表示できます。

Via the command `type`, we get the type of the Python object corresponding to  $\mathcal{M}$  (here the Python class `DifferentiableManifold_with_category`):

```
[7]: print(type(M))
```

```
<class  
'sage.manifolds.differentiable.manifold.DifferentiableManifold_with_category'>
```

`category()` 関数を使って、多様体  $\mathcal{M}$  のカテゴリーを調べます。この例では、この多様体のカテゴリーは  $\mathbb{R}$  の上の滑らかな多様体であることがわかります。

We can also ask for the category of  $\mathcal{M}$  and see that it is the category of smooth manifolds over  $\mathbb{R}$ :

```
[8]: category(M)
```

[8]: **Smooth $\mathbb{R}$**

多様体の指標 (を表す `generator`) は `irange()` メソッドで生成されます。この指標は `for`-ループなどで使われます。

The indices on the manifold are generated by the method `irange()`, to be used in loops:

```
[9]: [i for i in M.irange()]
```

[9]: [1, 2, 3]

このように単純な場合には、`list()` 関数が利用可能です。

```
[10]: list(M.irange())
```

[10]: [1, 2, 3]

`start_index` パラメータが明示的に指定されていない場合、多様体の指標は 0 からはじまります、例えばこの例のように (0,1,2) の範囲となります。

If the parameter `start_index` had not been specified, the default range of the indices would have been  $\{0, 1, 2\}$  instead:

```
[11]: M0 = Manifold(3, 'M', r'\mathcal{M}')
      [i for i in M0.irange()]
```

```
[11]: [0, 1, 2]
```

## 1.2 多様体上の座標系 (chart) の定義 : Defining a chart on the manifold

まずは、多様体  $M$  は一つの座標系で覆われると仮定します (そうでない場合は以下で議論されます)。座標系  $X$  は次のように宣言されます。

Let us assume that the manifold  $M$  can be covered by a single chart (other cases are discussed below); the chart is declared as follows:

```
[12]: X.<x,y,z> = M.chart()
```

上式の左辺にある `<x,y,z>` は Python の変数 `x`, `y` and `z` が座標系の軸名として使われることを意味します。これにより、座標軸を名前指定することが可能になります。

The writing `<x,y,z>` in the left-hand side means that the Python variables `x`, `y` and `z` are set to the three coordinates of the chart. This allows one to refer subsequently to the coordinates by their names.

この例では、関数 `chart()` は引数を持っていません。この場合には、座標の名前は、`x`, `y` と `z` (即ち、左辺の `<...>` 演算子で与えられた名前) になります。また座標の範囲は  $(-\infty, +\infty)$  となります。それ以外の座標の名前やその範囲を指定する場合には `chart()` にそれらを引数として与えなければなりません。座標名と異なる LaTeX で使われる名前を指定する場合も同様です。(以下にそのような例が示されます。)

In this example, the function `chart()` has no arguments, which implies that the coordinate symbols will be `x`, `y` and `z` (i.e. exactly the characters set in the `<...>` operator) and that each coordinate range is  $(-\infty, +\infty)$ . For other cases, an argument must be passed to `chart()` to specify the coordinate symbols and range, as well as the LaTeX symbol of a coordinate if the latter is different from the coordinate name (an example will be provided below).

座標系は座標系が覆う開集合 (ここでは多様体自身) と座標の組で表示されます。

The chart is displayed as a pair formed by the open set covered by it (here the whole manifold) and the coordinates:

```
[13]: print(X)
```

```
Chart (M, (x, y, z))
```

```
[14]: X
```

```
[14]: (M, (x, y, z))
```

それぞれの座標値は指標を使って入手できます。指標の指定方法は多様体の定義時の設定 (`start_index=1`) に従います。

The coordinates can be accessed individually, by means of their indices, following the convention defined by `start_index=1` in the manifold's definition:

```
[15]: X[1]
```

```
[15]: x
```

```
[16]: X[2]
```

```
[16]: y
```

```
[17]: X[3]
```

```
[17]: z
```

座標の全体は `[:]` 演算子を使って取得します。

The full set of coordinates is obtained by means of the operator `[:]`:

```
[18]: X[:]
```

```
[18]: (x, y, z)
```

座標系定義で使った `<x,y,z>` 演算子のおかげで、それぞれの座標をこの名前を使って直接取り扱えます。

Thanks to the operator `<x,y,z>` used in the chart declaration, each coordinate can be accessed directly via its name:

```
[19]: z is X[3]
```

```
[19]: True
```

座標は SageMath の記号表現になっています。

Coordinates are SageMath symbolic expressions:

```
[20]: print(type(z))
```

```
<class 'sage.symbolic.expression.Expression'>
```

### 1.2.1 座標の関数 : Functions of the chart coordinates

実数値をもつ座標の関数 (数学的にいえば、座標の終域で定義された関数) はチャートの `function()` メソッドを使って生成されます。

Real-valued functions of the chart coordinates (mathematically speaking, functions defined on the chart codomain) are generated via the method `function()` acting on the chart:

```
[21]: f = X.function(x+y^2+z^3)
      f
```

```
[21]: z3 + y2 + x
```

```
[22]: f.display()
```

```
[22]: (x, y, z) ↦ z3 + y2 + x
```

```
[23]: f(1, 2, 3)
```

```
[23]: 32
```

これらの関数は、SageManifolds のクラス `ChartFunction` (実際には、そのサブクラス `_h_.t_`) に属しています。

They belong to the class `ChartFunction` (actually the subclass `_h_.t_`):

```
[24]: print(type(f))
```

```
<class  
'sage.manifolds.chart_func.ChartFunctionRing_with_category.element_class'>
```

そしてこの `ChartFunction` クラスに属する関数は、標準的な SageMath の記号的関数とは全ての演算で自動的簡約化の振る舞いが異なります。

例えば、二つの記号的関数の和 :

and differ from SageMath standard symbolic functions by automatic simplifications in all operations. For instance, adding the two symbolic functions

```
[25]: f0(x,y,z) = cos(x)^2 ; g0(x,y,z) = sin(x)^2
```

は、次の結果になります。

results in

```
[26]: f0 + g0
```

```
[26]: (x, y, z) ↦ cos(x)2 + sin(x)2
```

一方、`ChartFunction` クラスに属する関数は自動的に簡約化されます。

while the sum of the corresponding functions in the class `ChartFunction` is automatically simplified:

```
[27]: f1 = X.function(cos(x)^2) ; g1 = X.function(sin(x)^2)  
f1 + g1
```

```
[27]: 1
```

同じ結果を記号的関数でも得るためには、`simplify_trig()` メソッドを使います。

To get the same output with symbolic functions, one has to invoke the method `simplify_trig()`:

```
[28]: (f0 + g0).simplify_trig()
```

```
[28]: (x, y, z) ↦ 1
```

違いは表示方法にもあります。記号関数 `f0` を表示させると、

Another difference regards the display; if we ask for the symbolic function `f0`, we get:

```
[29]: f0
```

```
[29]: (x, y, z) ↦ cos(x)2
```

となりますが、座標関数 `f1` を表示させると、座標表示だけが表示されます。

while if we ask for the chart function `f1`, we get only the coordinate expression:

```
[30]: f1
```

```
[30]: cos(x)2
```

`f0` の場合と同じような表示を望む場合には、`display()` メソッドを使います。

To get an output similar to that of `f0`, one should call the method `display()`:

```
[31]: f1.display()
```

```
[31]: (x, y, z) ↦ cos(x)2
```

`expr()` メソッドは基になる記号表現を返すことを覚えておいてください。

Note that the method `expr()` returns the underlying symbolic expression:

```
[32]: f1.expr()
```

```
[32]: cos(x)2
```

```
[33]: print(type(f1.expr()))
```

```
<class 'sage.symbolic.expression.Expression'>
```

### 1.2.2 多様体への二つ目の座標系の導入: Introducing a second chart on the manifold

まず、多様体  $\mathcal{M}$  の部分集合を考えましょう。例えば、 $\{y = 0, x \geq 0\}$  で定義される領域の補集合  $U$  です [  $(y \neq 0, x < 0)$  は  $y \neq 0$  OR  $x < 0$  を意味していることに注意してください;  $y \neq 0$  AND  $x < 0$  は  $[y \neq 0, x < 0]$  となります]。

Let us first consider an open subset of  $\mathcal{M}$ , for instance the complement  $U$  of the region defined by  $\{y = 0, x \geq 0\}$  (note that  $(y \neq 0, x < 0)$  stands for  $y \neq 0$  OR  $x < 0$ ; the condition  $y \neq 0$  AND  $x < 0$  would have been written  $[y \neq 0, x < 0]$  instead):

```
[34]: U = M.open_subset('U', coord_def={X: (y!=0, x<0)})
```

開部分集合  $U$  に制限された座標系  $X$  を  $X_U$  と呼ぶことにします。

Let us call  $X_U$  the restriction of the chart  $X$  to the open subset  $U$ :

```
[35]: X_U = X.restrict(U)
X_U
```

```
[35]: (U, (x, y, z))
```

$U$  上の別の座標系  $Y$  を球面座標  $(r, \theta, \phi)$  を使って導入します。

We introduce another chart on  $U$ , with spherical-type coordinates  $(r, \theta, \phi)$ :

```
[36]: Y.<r,th,ph> = U.chart(r'r:(0,+oo) th:(0,pi):\theta ph:(0,2*pi):\phi')
Y
```

```
[36]: (U, (r, \theta, \phi))
```

ここでの関数 `chart()` は引数を持っています。それは raw 文字列で LaTeX のシンボルを含んでいます。また、この文字列には座標の (既定値の  $(-\infty, +\infty)$  とは異なる) 範囲指定が含まれています。フィールドは `:` で区切られており、座標は空白文字で区切られています。座標  $r$  については、LaTeX シンボルを指定していないので、下限と上限の二つだけが指定されています。LaTeX シンボルは出力の際に使われます。

The function `chart()` has now some argument; it is a string, which contains specific LaTeX symbols, hence the prefix 'r' to it (for raw string). It also contains the coordinate ranges, since they are different from the default value, which is  $(-\infty, +\infty)$ . For a given coordinate, the various fields are separated by the character ':' and a space character separates the coordinates. Note that for the coordinate  $r$ , there are only two fields, since the LaTeX symbol has not to be specified. The LaTeX symbols are used for the outputs:

```
var('r1 ph1 th1') Y1.<r1,th1,ph1>= U.chart( coord_restrictions=[r1 > 0, th1 >= 0, th1 < pi, ph1 > 0, ph1 < 2*pi] ) Y1 # 接続/曲率など具体的な計算には、default_chart とつながる変換関数の定義が必要です。
transit_Y1_to_X = Y1.transition_map( X_U, [r1*sin(th1)*cos(ph1), r1*sin(th1)*sin(ph1), r1*cos(th1)] )
var('r2 ph2 th2') Y2= U.chart( names=("r2","th2:\theta_2","ph2:\phi_2"), coord_restrictions={((r2 > 0) , (0 <= th2 < pi) , (0 <= ph2 < 2*pi))} )
for c in ((r2 > 0) , (0 <= th2 < pi) , (0 <= ph2 < 2*pi)): assume(c) Y2
```

Sage が使っている座標の条件は、`assumptions()` メソッドを使って確認することができます。

The declared coordinate ranges are now known to Sage, as we may check by means of the command `assumptions()`:

```
[37]: assumptions()
```

```
[37]: [x is real, y is real, z is real, r is real, r > 0, th is real, \theta > 0, \theta < pi, ph is real, \phi > 0, \phi < 2*pi]
これらの条件は、表現の簡約化の際に使われます。
```

They are used in simplifications:

```
[38]: simplify(abs(r))
```

```
[38]: r
```

$r > 0$  と仮定されているので、`abs(r)` は  $r$  に簡約されます。

```
[39]: simplify(abs(x)) # no simplification occurs since x can take any value in R
```

```
[39]: |x|
```

$x$  は  $R$  の負の値を含むどんな値でも取ることができますから、`abs(x)` は簡約できません。



座標  $Y$  が宣言されたあとで、さらに 遷移写像を指定することで、 $Y$  の  $X_U$  への関係を完全に特定することができます。

After having been declared, the chart  $Y$  can be fully specified by its relation to the chart  $X_U$ , via a transition map:

```
[40]: transit_Y_to_X = Y.transition_map(
      X_U,
      [r*sin(th)*cos(ph), r*sin(th)*sin(ph), r*cos(th)]
    )
transit_Y_to_X
```

[40]:  $(U, (r, \theta, \phi)) \rightarrow (U, (x, y, z))$

```
[41]: transit_Y_to_X.display()
```

[41]: 
$$\begin{cases} x = r \cos(\phi) \sin(\theta) \\ y = r \sin(\phi) \sin(\theta) \\ z = r \cos(\theta) \end{cases}$$

`set_inverse()` メソッドを用いることで、遷移写像の逆写像を指定することができます。

The inverse of the transition map can be specified by means of the method `set_inverse()`:

```
[42]: transit_Y_to_X.set_inverse(sqrt(x^2+y^2+z^2), atan2(sqrt(x^2+y^2),z), atan2(y,
↳x))
```

Check of the inverse coordinate transformation:

```
r == r *passed*
th == arctan2(r*sin(th), r*cos(th)) **failed**
ph == arctan2(r*sin(ph)*sin(th), r*cos(ph)*sin(th)) **failed**
x == x *passed*
y == y *passed*
z == z *passed*
```

NB: a failed report can reflect a mere lack of simplification.

導かれた逆写像は、それと元の写像を組み合わせることで検証できます。左辺と右辺がそれぞれに対応しています。上の結果が示しているように、`th` と `ph` の検証が失敗したと報告されているのは、`arctan2` が関係する表式の単純化が不十分であることに起因しています。

逆写像を表示させてみましょう。

A check of the provided inverse is performed by composing it with the original transition map, on the left and on the right respectively. As indicated, the reported failure for `th` and `ph` is actually due to a lack of simplification of expressions involving `arctan2`. We have then

```
[43]: transit_Y_to_X.inverse().display()
```

[43]: 
$$\begin{cases} r = \sqrt{x^2 + y^2 + z^2} \\ \theta = \arctan\left(\sqrt{x^2 + y^2}, z\right) \\ \phi = \arctan(y, x) \end{cases}$$

この段階で、多様体のアトラス ( `user atlas` であって最大アトラスではありません) は三つの座標系を含んでいます。

At this stage, the manifold's atlas (the “user atlas”, not the maximal atlas!) contains three charts:

```
[44]: M.atlas()
```

```
[44]: [(M, (x, y, z)), (U, (x, y, z)), (U, (r, theta, phi))]
```

多様体に最初に定義された座標系は既定の座標系とみなされます。既定座標系はあとで、`set_default_chart()` メソッドを使って変更できます。

The first chart defined on the manifold is considered as the manifold's default chart (it can be changed by the method `set_default_chart()`):

```
[45]: M.default_chart()
```

```
[45]: (M, (x, y, z))
```

各々の開部分空間は各自のアトラスを持っています。開部分空間はそれ自身でも多様体であるからです。

Each open subset has its own atlas (since an open subset of a manifold is a manifold by itself):

```
[46]: U.atlas()
```

```
[46]: [(U, (x, y, z)), (U, (r, theta, phi))]
```

```
[47]: U.default_chart()
```

```
[47]: (U, (x, y, z))
```

```
[48]: Y.plot(X, color={r:'red', th:'green', ph:'blue'})
```

```
[48]: Graphics3d Object
```

`Y.plot(X)` コマンドを使って、座標系  $Y$  を座標系  $X$  を使って表示することができます。このコマンドは、座標系  $Y$  で一定の座標値を持つ線を、座標系  $X$  に基づく”直交フレーム”中に表示します。

We can draw the chart  $Y$  in terms of the chart  $X$  via the command `Y.plot(X)`, which shows the lines of constant coordinates from the  $Y$  chart in a “Cartesian frame” based on the  $X$  coordinates:

```
[49]: Y.plot(X, color={r:'blue', th:'green', ph:'red'}, aspect_ratio=1)
```

```
[49]: Graphics3d Object
```

コマンド `plot()` は多くのオプションを持っています。オプションを使い、描画される座標線の数、スタイル、色、そして座標の範囲などを指定します。(参照:[list of all options](#)):

The command `plot()` allows for many options, to control the number of coordinate lines to be drawn, their style and color, as well as the coordinate ranges (cf. the [list of all options](#)):

```
[50]: Y.plot(X, ranges={r:(1,2), th:(0,pi/2)}, number_values=4,
        color={r:'blue', th:'green', ph:'red'}, aspect_ratio=1)
```

[50]: Graphics3d Object

逆に、(制限された) 座標系  $X|_U$  を座標系  $Y$  を使って表示することも可能です。この場合、座標系  $X$  の全てを表示することはできません、なぜなら  $X$  の定義域(ドメイン)は  $Y$  のそれより広いからです。

Conversly, the chart  $X|_U$  can be plotted in terms of the chart  $Y$  (this is not possible for the whole chart  $X$  since its domain is larger than that of chart  $Y$ ):

```
[51]: graph = X_U.plot(Y,color={x:'blue',y:'green', z:'red'})
      show(graph, axes_labels=['r','theta','phi'])
```

Graphics3d Object

オプションを追加してすることで、グラフを見やすくできます。

```
[52]: graph = X_U.plot(Y, ranges={x:(0,2), y:(0,2),z:(0,2),
        r:(0,5), th:(0,pi/2), ph:(0, 2*pi) },
        number_values=4,
        color={x:'blue',y:'green', z:'red'})
      show(graph, axes_labels=['r','theta','phi'])
```

Graphics3d Object

### 1.3 多様体上の点 :Points on the manifold

多様体  $\mathcal{M}$  上の点は、与えられた座標系での座標を使って定義されます。

A point on  $\mathcal{M}$  is defined by its coordinates in a given chart:

```
[53]: p = M.point((1,2,-1), chart=X, name='p') ; print(p) ; p
```

Point  $p$  on the 3-dimensional differentiable manifold  $M$

[53]:  $p$

$X = (\mathcal{M}, (x, y, z))$  は既定の座標系ですから、その名前を省略することが可能です。

Since  $X = (\mathcal{M}, (x, y, z))$  is the manifold's default chart, its name can be omitted:

```
[54]: p = M.point((1,2,-1), name='p') ; print(p) ; p
```

Point  $p$  on the 3-dimensional differentiable manifold  $M$

[54]:  $p$

もちろん  $p$  は多様体  $\mathcal{M}$  中の点です。

Of course,  $p$  belongs to  $\mathcal{M}$ :

```
[55]: p in M
```

```
[55]: True
```

この点は  $U$  の中の点でもあります。

It is also in  $U$ :

```
[56]: p in U
```

```
[56]: True
```

実際、点  $p$  の座標  $y$  は  $0$  ではありません ( $y \neq 0$ ): ( $U$  は  $y \neq 0 \parallel x < 0$  で定義された開領域)

Indeed the coordinates of  $p$  have  $y \neq 0$ :

```
[57]: p.coord(X)
```

```
[57]: (1, 2, -1)
```

ここで、 $X$  は  $M$  の既定座標系であることから `coord()` にわたす引数では、省略可能であることに注意してください。

Note in passing that since  $X$  is the default chart on  $M$ , its name can be omitted in the arguments of `coord()`:

```
[58]: p.coord()
```

```
[58]: (1, 2, -1)
```

$p$  の座標値は、座標系を点  $p$  に作用させることによっても得ることができます (座標系の定義により!):

The coordinates of  $p$  can also be obtained by letting the chart acting of the point (from the very definition of a chart!):

```
[59]: X(p)
```

```
[59]: (1, 2, -1)
```

別の点  $q$  を  $y = 0$  and  $x \geq 0$  を満たす一つの点として定義します。

Let  $q$  be a point with  $y = 0$  and  $x \geq 0$ :

```
[60]: q = M.point((1, 0, 2), name='q')
```

$q$  は  $U$  には属しません。

This time, the point does not belong to  $U$ :

```
[61]: q in U
```

```
[61]: False
```

従って、点  $q$  の座標系  $Y = (U, (r, \theta, \phi))$  での座標値を求めることはできません。

Accordingly, we cannot ask for the coordinates of  $q$  in the chart  $Y = (U, (r, \theta, \phi))$ :

```
[62]: try:
      q.coord(Y)
      except ValueError as exc:
          print("Error: " + str(exc))
```

Error: the point does not belong to the domain of Chart (U, (r, th, ph))

しかし、点  $p$  については可能です。

but we can for point  $p$ :

```
[63]: p.coord(Y)
```

```
[63]: ( $\sqrt{3}\sqrt{2}, \pi - \arctan(\sqrt{5}), \arctan(2)$ )
```

```
[64]: Y(p)
```

```
[64]: ( $\sqrt{3}\sqrt{2}, \pi - \arctan(\sqrt{5}), \arctan(2)$ )
```

点は比較可能です。 :

Points can be compared:

```
[65]: q == p
```

```
[65]: False
```

```
[66]: p1 = U.point((sqrt(3)*sqrt(2), pi-atan(sqrt(5)), atan(2)), chart=Y)
      p1 == p
```

```
[66]: True
```

SageMath の用語では、点は それが定義された多様体を 親 (**parents**) として持つ 要素 (**elements**) です。

In SageMath's terminology, points are elements, whose parents are the manifold on which they have been defined:

```
[67]: p.parent()
```

```
[67]:  $\mathcal{M}$ 
```

```
[68]: q.parent()
```

```
[68]:  $\mathcal{M}$ 
```

```
[69]: p1.parent()
```

```
[69]:  $U$ 
```

## 1.4 スカラー場 : Scalar fields

スカラー場は微分可能な写像  $U \rightarrow \mathbb{R}$  のことです。ここで  $U$  は多様体  $\mathcal{M}$  の開部分集合です。スカラー場はその定義域を覆う座標系を使った式で定義されます。一般には、一つ以上の座標系が定義域全体を覆うためには必要となります。

A **scalar field** is a differentiable map  $U \rightarrow \mathbb{R}$ , where  $U$  is an open subset of  $\mathcal{M}$ .

A scalar field is defined by its expressions in terms of charts covering its domain (in general more than one chart is necessary to cover all the domain):

```
[70]: f = U.scalar_field({X_U: x+y^2+z^3}, name='f') ; print(f)
```

Scalar field  $f$  on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

座標で書かれた表式は、座標系をキーとした Python の辞書型データ  $\{X_U: x+y^2+z^3\}$  として渡されます。この例では、辞書には一つの座標系しかありませんので、次のような表現も可能です。

The coordinate expressions of the scalar field are passed as a Python dictionary, with the charts as keys, hence the writing  $\{X_U: x+y^2+z^3\}$ .

Since in the present case, there is only one chart in the dictionary, an alternative writing is

```
[71]: f = U.scalar_field({X_U:x+y^2+z^3}, name='f')
      print(f)
```

Scalar field  $f$  on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

$X_U$  は定義域 ( $U$ ) の既定の座標系ですので、上記の表現では省略可能です。

Since  $X_U$  is the domain's default chart, it can be omitted in the above declaration:

```
[72]: f = U.scalar_field(x+y^2+z^3, name='f') ; print(f)
```

Scalar field  $f$  on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

スカラー場は定義域となる多様体から実数体への写像ですから、それは点に対して働き、座標値に対して働くものではありません。

As a mapping  $U \subset \mathcal{M} \rightarrow \mathbb{R}$ , a scalar field acts on points, not on coordinates:

```
[73]: f(p)
```

```
[73]: 4
```

スカラー場の `display()` メソッドは与えられた座標系でのスカラー場の表現を表示します。

The method `display()` provides the expression of the scalar field in terms of a given chart:

```
[74]: f.display(X_U)
```

```
[74]:
```

$$f: U \longrightarrow \mathbb{R}$$

$$(x, y, z) \longmapsto z^3 + y^2 + x$$

引数なしで呼び出された場合、`display()` メソッドは 定義域  $U$  の上で定義された全ての座標系について、座標表現を表示します。(但し副座標系, 即ち制限された副定義域のみで定義された座標系は除きます。)

If no argument is provided, the method `display()` shows the coordinate expression of the scalar field in all the charts defined on the domain (except for *subcharts*, i.e. the restrictions of some chart to a subdomain):

[75]: `f.display()`

[75]:  $f: U \longrightarrow \mathbb{R}$

$$(x, y, z) \longmapsto z^3 + y^2 + x$$

$$(r, \theta, \phi) \longmapsto r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r \cos(\phi) \sin(\theta)$$

スカラー場  $f$  の 座標  $(r, \theta, \phi)$  での表現はユーザによって与えられたものではなく、座標系の遷移写像で定義された座標変換のルールを使って自動的に導出されていることに注意してください。

Note that the expression of  $f$  in terms of the coordinates  $(r, \theta, \phi)$  has not been provided by the user but has been automatically computed by means of the change-of-coordinate formula declared above in the transition map.

[76]: `f.display(Y)`

[76]:  $f: U \longrightarrow \mathbb{R}$

$$(r, \theta, \phi) \longmapsto r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r \cos(\phi) \sin(\theta)$$

各座標系で、スカラー場は座標の関数 (`CoordFunctionSymb` のオブジェクト) として表現されます。このオブジェクトは `coord_function()` メソッドを通じてアクセス可能です。

In each chart, the scalar field is represented by a function of the chart coordinates (an object of the type `CoordFunctionSymb` described above), which is accessible via the method `coord_function()`:

[77]: `f.coord_function(X_U)`

[77]:  $z^3 + y^2 + x$

[78]: `f.coord_function(X_U).display()`

[78]:  $(x, y, z) \mapsto z^3 + y^2 + x$

[79]: `f.coord_function(Y)`

[79]:  $r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r \cos(\phi) \sin(\theta)$

[80]: `f.coord_function(Y).display()`

[80]:  $(r, \theta, \phi) \mapsto r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r \cos(\phi) \sin(\theta)$

“生” の記号的表現は、`expr()` メソッドの戻り値として返されます。

The “raw” symbolic expression is returned by the method `expr()`:

```
[81]: f.expr(X_U)
```

```
[81]: z3 + y2 + x
```

```
[82]: f.expr(Y)
```

```
[82]: r3 cos(θ)3 + r2 sin(φ)2 sin(θ)2 + r cos(φ) sin(θ)
```

```
[83]: f.expr(Y) is f.coord_function(Y).expr()
```

```
[83]: True
```

スカラー場は具体的な式ではなく、座標値の関数を用いて定義することも可能です。

A scalar field can also be defined by some unspecified function of the coordinates:

```
[84]: h = U.scalar_field(function('H')(x, y, z), name='h') ; print(h)
```

Scalar field `h` on the Open subset `U` of the 3-dimensional differentiable manifold `M`

```
[85]: h.display()
```

```
[85]: h: U      → ℝ
      (x, y, z) ↦ H(x, y, z)
      (r, θ, φ) ↦ H(r cos(φ) sin(θ), r sin(φ) sin(θ), r cos(θ))
```

```
[86]: h.display(Y)
```

```
[86]: h: U      → ℝ
      (r, θ, φ) ↦ H(r cos(φ) sin(θ), r sin(φ) sin(θ), r cos(θ))
```

```
[87]: h(p) # remember that p is the point of coordinates (1,2,-1) in the chart X_U
```

```
[87]: H(1,2,-1)
```

$f$  の親 (parent) は すべての  $U$  上の滑らかなスカラー場からなる集合  $C^\infty(U)$  です。これは  $\mathbb{R}$  上の可換代数です。

The parent of  $f$  is the set  $C^\infty(U)$  of all smooth scalar fields on  $U$ , which is a commutative algebra over  $\mathbb{R}$ :

```
[88]: CU = f.parent() ; CU
```

```
[88]: C∞(U)
```

```
[89]: print(CU)
```

Algebra of differentiable scalar fields on the Open subset `U` of the 3-dimensional differentiable manifold `M`



```
[90]: CU.category()
```

[90]: **JoinCategory**

代数  $CU$  の基礎体となる環は実数体  $\mathbb{R}$  です。それは SageMath では Symbolic Ring (SR) で表現されます。

The base ring of the algebra is the field  $\mathbb{R}$ , which is represented here by SageMath's Symbolic Ring (SR):

```
[91]: CU.base_ring()
```

[91]: SR

スカラー場の演算は代数構造を通じて定義されます。

Arithmetic operations on scalar fields are defined through the algebra structure:

```
[92]: s = f + 2*h ; print(s)
```

Scalar field on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

```
[93]: s.display()
```

[93]:

$$\begin{aligned} U &\longrightarrow \mathbb{R} \\ (x, y, z) &\longmapsto z^3 + y^2 + x + 2H(x, y, z) \\ (r, \theta, \phi) &\longmapsto r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r \cos(\phi) \sin(\theta) + 2H(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \end{aligned}$$

## 1.5 接空間 : Tangent spaces

点  $p$  での多様体の接ベクトル空間を次のようにして取り出します。

The tangent vector space to the manifold at point  $p$  is obtained as follows:

```
[94]: Tp = M.tangent_space(p) ; Tp
```

[94]:  $T_p \mathcal{M}$

```
[95]: print(Tp)
```

Tangent space at Point  $p$  on the 3-dimensional differentiable manifold  $M$

$T_p \mathcal{M}$  は  $\mathbb{R}$  上の 3次元のベクトル空間です。

$T_p \mathcal{M}$  is a 3-dimensional vector space over  $\mathbb{R}$  (represented here by SageMath's Symbolic Ring (SR))  
:

```
[96]: print(Tp.category())
```

Category of finite dimensional vector spaces over Symbolic Ring

```
[97]: Tp.dim()
```

[97]:

3

$T_p \mathcal{M}$  のベクトルの基底は点  $p$  の周りのベクトルフレームから誘導されます。

$T_p \mathcal{M}$  is automatically endowed with vector bases deduced from the vector frames defined around the point:

```
[98]: Tp.bases()
```

```
[98]:  $\left[ \left( \frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right), \left( \frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi} \right) \right]$ 
```

対照的に、点  $q$  については、あらかじめ定義された基底は一つだけです。  $q$  はドメイン  $U$  に含まれませんので、  $(r, \theta, \phi)$  に関連づけられていません。

For the tangent space at the point  $q$ , on the contrary, there is only one pre-defined basis, since  $q$  is not in the domain  $U$  of the frame associated with coordinates  $(r, \theta, \phi)$ :

```
[99]: Tq = M.tangent_space(q)
Tq.bases()
```

```
[99]:  $\left[ \left( \frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right) \right]$ 
```

ランダムな要素 :

A random element:

```
[100]: v = Tp.an_element() ; print(v)
```

Tangent vector at Point p on the 3-dimensional differentiable manifold M

```
[101]: v.display()
```

```
[101]:  $\frac{\partial}{\partial x} + 2\frac{\partial}{\partial y} + 3\frac{\partial}{\partial z}$ 
```

```
[102]: u = Tq.an_element() ; print(u)
```

Tangent vector at Point q on the 3-dimensional differentiable manifold M

```
[103]: u.display()
```

```
[103]:  $\frac{\partial}{\partial x} + 2\frac{\partial}{\partial y} + 3\frac{\partial}{\partial z}$ 
```

上記の簡略化された表現 (基底ベクトルに点  $p$  あるいは  $q$  の指定がない) から類推されるのとは異なり、  $u$  と  $v$  は異なるベクトルです。これらは異なったベクトル空間に属しています。:

Note that, despite what the above simplified writing may suggest (the mention of the point  $p$  or  $q$  is omitted in the basis vectors),  $u$  and  $v$  are different vectors, for they belong to different vector spaces:

```
[104]: v.parent()
```

[104]:  $T_p \mathcal{M}$

```
[105]: u.parent()
```

[105]:  $T_q \mathcal{M}$

とくに  $u$  と  $v$  を足し合わせることはできません。

In particular, it is not possible to add  $u$  and  $v$ :

一般に二つの点のベクトルの差分として、共変微分の積分を考えても、それは積分の経路に依存することになるので、一意に定義することはできない。

```
[106]: try:
        s = u + v
    except TypeError as exc:
        print("Error: ", str(exc))
```

```
Error: unsupported operand parent(s) for +: 'Tangent space at Point q on the
3-dimensional differentiable manifold M' and 'Tangent space at Point p on the
3-dimensional differentiable manifold M'
```

## 1.6 ベクトル場 : Vector Fields

全ての座標系は座標系のドメインの上のベクトルフレームを持っています。これは、coordinate basis と呼ばれます、

Each chart defines a vector frame on the chart domain: the so-called coordinate basis:

```
[107]: X.frame()
```

[107]:  $\left( \mathcal{M}, \left( \frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right) \right)$

```
[108]: X.frame().domain() # this frame is defined on the whole manifold このフレームは
多様体全体で定義されています。
```

[108]:  $\mathcal{M}$

```
[109]: Y.frame()
```

[109]:  $\left( U, \left( \frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi} \right) \right)$

座標系  $Y$  のフレームは、部分多様体  $U$  でのみ定義されています。

# this frame is defined only on

```
[110]: Y.frame().domain() # このフレームは部分多様体  $U$ でのみ定義されています。
```

[110]:  $U$

frames() メソッドによって、与えられた開空間で定義されたフレームのリストが入手可能です。

The list of frames defined on a given open subset is returned by the method frames():

```
[111]: M.frames()
```

```
[111]:  $\left[ \left( \mathcal{M}, \left( \frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right) \right), \left( U, \left( \frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right) \right), \left( U, \left( \frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi} \right) \right) \right]$ 
```

```
[112]: U.frames()
```

```
[112]:  $\left[ \left( U, \left( \frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right) \right), \left( U, \left( \frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi} \right) \right) \right]$ 
```

```
[113]: M.default_frame()
```

```
[113]:  $\left( \mathcal{M}, \left( \frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right) \right)$ 
```

set\_default\_frame() メソッドで別途明示的に指定されない限り、基底の座標系に関連づけられるベクトルフレームが基底のフレームです。

Unless otherwise specified (via the command set\_default\_frame()), the default frame is that associated with the default chart:

```
[114]: M.default_frame() is M.default_chart().frame()
```

```
[114]: True
```

```
[115]: U.default_frame() is U.default_chart().frame()
```

```
[115]: True
```

フレームのそれぞれの要素は指標を使って指定できます。

Individual elements of a frame can be accessed by means of their indices:

```
[116]: e = U.default_frame() ; e2 = e[2] ; e2
```

```
[116]:  $\frac{\partial}{\partial y}$ 
```

```
[117]: print(e2)
```

Vector field  $\partial / \partial y$  on the Open subset U of the 3-dimensional differentiable manifold M

新しいベクトル場を次のようにして、定義できます。

We may define a new vector field as follows:

```
[118]: v = e[2] + 2*x*e[3] ; print(v)
```

Vector field on the Open subset U of the 3-dimensional differentiable manifold M

```
[119]: v.display()
```

```
[119]:  $\frac{\partial}{\partial y} + 2x \frac{\partial}{\partial z}$ 
```

ベクトル場は与えられたベクトルフレームに関する要素を指定することで、定義できます。もしベクトルフレームが指定されていない場合には、既定のフレームがもちろん仮定されます。

A vector field can be defined by its components with respect to a given vector frame. When the latter is not specified, the open set's default frame is of course assumed:

```
[120]: v = U.vector_field(name='v') # vector field defined on the open set U
v[1] = 1+y
v[2] = -x
v[3] = x*y*z
v.display()
```

```
[120]:  $v = (y + 1) \frac{\partial}{\partial x} - x \frac{\partial}{\partial y} + xyz \frac{\partial}{\partial z}$ 
```

SageMath 8.8 以降では、ベクトルフィールドの成分はそれを宣言する時に設定することができます。従って、上記は次の宣言と等価です。

Since version 8.8 of SageMath, it is possible to initialize the components of the vector field while declaring it, so that the above is equivalent to

```
[121]: v = U.vector_field(1+y, -x, x*y*z, name='v') # valid only in SageMath 8.8 and
↳higher
v.display()
```

```
[121]:  $v = (y + 1) \frac{\partial}{\partial x} - x \frac{\partial}{\partial y} + xyz \frac{\partial}{\partial z}$ 
```

$U$  上のベクトル場は Sage の **要素 (element)** オブジェクトです。その **親 (parent)** は  $U$  上で定義されたベクトル場の集合  $\mathfrak{X}(U)$  に設定されています。

Vector fields on  $U$  are Sage element objects, whose parent is the set  $\mathfrak{X}(U)$  of vector fields defined on  $U$ :

```
[122]: v.parent()
```

```
[122]:  $\mathfrak{X}(U)$ 
```

集合  $\mathfrak{X}(U)$  は  $U$  で定義されたのスカラール場の可換代数  $C^\infty(U)$  上のモジュールです。

The set  $\mathfrak{X}(U)$  is a module over the commutative algebra  $C^\infty(U)$  of scalar fields on  $U$ :

```
[123]: print(v.parent())
```

```
Free module X(U) of vector fields on the Open subset U of the 3-dimensional
differentiable manifold M
```

```
[124]: print(v.parent().category())
```

Category of finite dimensional modules over Algebra of differentiable scalar fields on the Open subset U of the 3-dimensional differentiable manifold M

```
[125]: v.parent().base_ring()
```

[125]:  $C^\infty(U)$

ベクトル場はスカラー場に作用します。

A vector field acts on scalar fields:

```
[126]: f.display()
```

[126]:  $f: U \rightarrow \mathbb{R}$   
 $(x, y, z) \mapsto z^3 + y^2 + x$   
 $(r, \theta, \phi) \mapsto r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r \cos(\phi) \sin(\theta)$

```
[127]: s = v(f) ; print(s)
```

Scalar field v(f) on the Open subset U of the 3-dimensional differentiable manifold M

```
[128]: s.display()
```

[128]:  $v(f): U \rightarrow \mathbb{R}$   
 $(x, y, z) \mapsto 3xyz^3 - (2x - 1)y + 1$   
 $(r, \theta, \phi) \mapsto r \sin(\phi) \sin(\theta) + \left(3r^5 \cos(\phi) \cos(\theta)^3 \sin(\phi) - 2r^2 \cos(\phi) \sin(\phi)\right) \sin(\theta)^2 + 1$

```
[129]: e[3].display()
```

[129]:  $\frac{\partial}{\partial z} = \frac{\partial}{\partial z}$

```
[130]: e[3](f).display()
```

[130]:  $\frac{\partial}{\partial z}(f): U \rightarrow \mathbb{R}$   
 $(x, y, z) \mapsto 3z^2$   
 $(r, \theta, \phi) \mapsto 3r^2 \cos(\theta)^2$

設定されていない要素は、0 と仮定されます。

Unset components are assumed to be zero:

```
[131]: w = U.vector_field(name='w')
w[2] = 3
w.display()
```

[131]:  $w = 3 \frac{\partial}{\partial y}$

U 上のベクトル場は座標系  $(r, \theta, \phi)$  に関係付けられているベクトルフレームで展開することができます。

A vector field on  $U$  can be expanded in the vector frame associated with the chart  $(r, \theta, \phi)$ :

[132]: `v.display(Y.frame())`

[132]: 
$$v = \left( \frac{xyz^2 + x}{\sqrt{x^2 + y^2 + z^2}} \right) \frac{\partial}{\partial r} + \left( -\frac{(x^3y + xy^3 - x)\sqrt{x^2 + y^2}z}{x^4 + 2x^2y^2 + y^4 + (x^2 + y^2)z^2} \right) \frac{\partial}{\partial \theta} + \left( -\frac{x^2 + y^2 + y}{x^2 + y^2} \right) \frac{\partial}{\partial \phi}$$

既定の設定では、要素は既定の座標  $(x, y, z)$  を用いて、表現されます。これらの要素を別の座標  $(r, \theta, \phi)$  を用いて表現するには、`display()` メソッドの2番目の引数として、対応する座標系を指定することが必要です。

By default, the components are expressed in terms of the default coordinates  $(x, y, z)$ . To express them in terms of the coordinates  $(r, \theta, \phi)$ , one should add the corresponding chart as the second argument of the method `display()`:

[133]: `v.display(Y.frame(), Y)`

[133]: 
$$v = \left( r^3 \cos(\phi) \cos(\theta)^2 \sin(\phi) \sin(\theta)^2 + \cos(\phi) \sin(\theta) \right) \frac{\partial}{\partial r} + \left( -\frac{r^3 \cos(\phi) \cos(\theta) \sin(\phi) \sin(\theta)^3 - \cos(\phi) \cos(\theta)}{r} \right) \frac{\partial}{\partial \theta} + \left( -\frac{r \sin(\theta) + \sin(\phi)}{r \sin(\theta)} \right) \frac{\partial}{\partial \phi}$$

[134]: `for i in M.irange():  
show(e[i].display(Y.frame(), Y))`

$$\frac{\partial}{\partial x} = \cos(\phi) \sin(\theta) \frac{\partial}{\partial r} + \frac{\cos(\phi) \cos(\theta)}{r} \frac{\partial}{\partial \theta} - \frac{\sin(\phi)}{r \sin(\theta)} \frac{\partial}{\partial \phi}$$

$$\frac{\partial}{\partial y} = \sin(\phi) \sin(\theta) \frac{\partial}{\partial r} + \frac{\cos(\theta) \sin(\phi)}{r} \frac{\partial}{\partial \theta} + \frac{\cos(\phi)}{r \sin(\theta)} \frac{\partial}{\partial \phi}$$

$$\frac{\partial}{\partial z} = \cos(\theta) \frac{\partial}{\partial r} - \frac{\sin(\theta)}{r} \frac{\partial}{\partial \theta}$$

既定のフレームに関するテンソル場の要素は、コマンド `[:]` を使うことで、リストとして取り出せます。

The components of a tensor field w.r.t. the default frame can also be obtained as a list, via the command `[:]`:

[135]: `v[:]`

[135]:  $[y + 1, -x, xyz]$

別の方法として、`display_comp()` メソッドを使うこともできます。

An alternative is to use the method `display_comp()`:

[136]: `v.display_comp()`

[136]: 
$$\begin{aligned} v^x &= y + 1 \\ v^y &= -x \\ v^z &= xyz \end{aligned}$$

別のフレームについての要素を取り出すには、`comp()` メソッドを通じて要素を取り出すフレームを指定する必要があります。

To obtain the components w.r.t. to another frame, one may go through the method `comp()` and specify the frame:

```
[137]: v.comp(Y.frame())[:]
```

$$[137]: \left[ \frac{xyz^2 + x}{\sqrt{x^2 + y^2 + z^2}}, -\frac{(x^3y + xy^3 - x)\sqrt{x^2 + y^2}z}{x^4 + 2x^2y^2 + y^4 + (x^2 + y^2)z^2}, -\frac{x^2 + y^2 + y}{x^2 + y^2} \right]$$

しかしながら、最初の引数にフレームを与える 短縮表現も用意されています。

However a shortcut is to provide the frame as the first argument of the square brackets:

```
[138]: v[Y.frame(), :]
```

$$[138]: \left[ \frac{xyz^2 + x}{\sqrt{x^2 + y^2 + z^2}}, -\frac{(x^3y + xy^3 - x)\sqrt{x^2 + y^2}z}{x^4 + 2x^2y^2 + y^4 + (x^2 + y^2)z^2}, -\frac{x^2 + y^2 + y}{x^2 + y^2} \right]$$

```
[139]: v.display_comp(Y.frame())
```

$$[139]: \begin{aligned} v^r &= \frac{xyz^2 + x}{\sqrt{x^2 + y^2 + z^2}} \\ v^\theta &= -\frac{(x^3y + xy^3 - x)\sqrt{x^2 + y^2}z}{x^4 + 2x^2y^2 + y^4 + (x^2 + y^2)z^2} \\ v^\phi &= -\frac{x^2 + y^2 + y}{x^2 + y^2} \end{aligned}$$

要素は、既定の座標系の座標で表現されます。これらの別の座標系の座標、例えば  $(r, \theta, \phi)$ 、で表現したい時には、`[]` の最後の引数として、座標系の名前を与えます。

Components are shown expressed in terms of the default's coordinates; to get them in terms of the coordinates  $(r, \theta, \phi)$  instead, add the chart name as the last argument in the square brackets:

```
[140]: v[Y.frame(), :, Y]
```

$$[140]: \left[ r^3 \cos(\phi) \cos(\theta)^2 \sin(\phi) \sin(\theta)^2 + \cos(\phi) \sin(\theta), -\frac{r^3 \cos(\phi) \cos(\theta) \sin(\phi) \sin(\theta)^3 - \cos(\phi) \cos(\theta)}{r}, -\frac{r \sin(\theta) + \sin(\phi)}{r \sin(\theta)} \right]$$

あるいは、`display_comp()` の引数に座標系の名前を与えます。

or specify the chart in `display_comp()`:

```
[141]: v.display_comp(Y.frame(), chart=Y)
```

$$[141]: \begin{aligned} v^r &= r^3 \cos(\phi) \cos(\theta)^2 \sin(\phi) \sin(\theta)^2 + \cos(\phi) \sin(\theta) \\ v^\theta &= -\frac{r^3 \cos(\phi) \cos(\theta) \sin(\phi) \sin(\theta)^3 - \cos(\phi) \cos(\theta)}{r} \\ v^\phi &= -\frac{r \sin(\theta) + \sin(\phi)}{r \sin(\theta)} \end{aligned}$$

```
[142]: v.display_comp(Y.frame(), chart=X_U)
```

```
[142]:
```



$$\begin{aligned}
v^r &= \frac{xyz^2+x}{\sqrt{x^2+y^2+z^2}} \\
v^\theta &= -\frac{(x^3y+xy^3-x)\sqrt{x^2+y^2}z}{x^4+2x^2y^2+y^4+(x^2+y^2)z^2} \\
v^\phi &= -\frac{x^2+y^2+y}{x^2+y^2}
\end{aligned}$$

ベクトル場の成分を座標についての表現ではなく、スカラーフィールドとして取り出すために、二重の括弧 [[]] を使います。

To get some vector component as a scalar field instead of a coordinate expression, use double square brackets:

```
[143]: print(v[[1]])
```

Scalar field on the Open subset U of the 3-dimensional differentiable manifold M

```
[144]: v[[1]].display()
```

```
[144]: U      -> R
(x, y, z)  -> y + 1
(r, theta, phi) -> r sin(phi) sin(theta) + 1
```

```
[145]: [v[[i]].expr(X_U) for i in range(1,4)]
```

```
[145]: [y + 1, -x, xyz]
```

ベクトル場は関数オブジェクトを要素として定義することも可能です。

A vector field can be defined with components being unspecified functions of the coordinates:

```
[146]: u = U.vector_field(name='u')
u[:] = [function('u_x')(x,y,z),
        function('u_y')(x,y,z),
        function('u_z')(x,y,z)]
u.display()
```

```
[146]: u = u_x(x, y, z) * partial_x + u_y(x, y, z) * partial_y + u_z(x, y, z) * partial_z
```

```
[147]: s = v + u ; s.set_name('s') ; s.display()
```

```
[147]: s = (y + u_x(x, y, z) + 1) * partial_x + (-x + u_y(x, y, z)) * partial_y + (xyz + u_z(x, y, z)) * partial_z
```

### 1.6.1 ある点でのベクトル場の値 : Values of vector fields at a given point

多様体中のある点でのベクトル場の値は `at()` メソッドを通じて入手します。

The value of a vector field at some point of the manifold is obtained via the method `at()`:

```
[148]: vp = v.at(p) ; print(vp)
```

Tangent vector v at Point p on the 3-dimensional differentiable manifold M

```
[149]: vp.display()
```

```
[149]: 
$$v = 3\frac{\partial}{\partial x} - \frac{\partial}{\partial y} - 2\frac{\partial}{\partial z}$$

```

座標系  $X_U = (x, y, z)$  を使った、点  $p$  の座標と、ベクトル場  $v$  の要素は次のようになっていたことを思い出してください。

Indeed, recall that, w.r.t. chart  $X_U=(x, y, z)$ , the coordinates of the point  $p$  and the components of the vector field  $v$  are

```
[150]: p.coord(X_U)
```

```
[150]: (1, 2, -1)
```

```
[151]: v.display(X_U.frame(), X_U)
```

```
[151]: 
$$v = (y + 1)\frac{\partial}{\partial x} - x\frac{\partial}{\partial y} + xyz\frac{\partial}{\partial z}$$

```

記述を簡単にするために、点  $p$  でのベクトル場の値とベクトル場は同じ記号 ( $v$ ) を使って表示されています。set\_name() メソッドを使うことで、表示に使われる記号を変更できます。

Note that to simplify the writing, the symbol used to denote the value of the vector field at point  $p$  is the same as that of the vector field itself (namely  $v$ ); this can be changed by the method set\_name():

```
[152]: vp.set_name(latex_name='v|_p')
vp.display()
```

```
[152]: 
$$v|_p = 3\frac{\partial}{\partial x} - \frac{\partial}{\partial y} - 2\frac{\partial}{\partial z}$$

```

もちろん、 $v|_p$  は  $p$  の接ベクトル空間に属しています。

Of course,  $v|_p$  belongs to the tangent space at  $p$ :

```
[153]: vp.parent()
```

```
[153]:  $T_p \mathcal{M}$ 
```

```
[154]: vp in M.tangent_space(p)
```

```
[154]: True
```

```
[155]: up = u.at(p) ; print(up)
```

Tangent vector  $u$  at Point  $p$  on the 3-dimensional differentiable manifold  $M$

```
[156]: up.display()
```

```
[156]: 
$$u = u_x(1, 2, -1)\frac{\partial}{\partial x} + u_y(1, 2, -1)\frac{\partial}{\partial y} + u_z(1, 2, -1)\frac{\partial}{\partial z}$$

```

## 1.7 1-形式: 1-forms

$\mathcal{M}$  の 1-形式とは、線形形式の場です。例えば、スカラー場の微分は 1-形式です。

A 1-form on  $\mathcal{M}$  is a field of linear forms. For instance, it can be the differential of a scalar field:

```
[157]: df = f.differential() ; print(df)
```

1-form df on the Open subset U of the 3-dimensional differentiable manifold M

```
[158]: df.display()
```

```
[158]: df = dx + 2ydy + 3z2dz
```

上記の表現では、1-形式は座標系  $(x, y, z)$  から導かれる基底  $(dx, dy, dz)$  で展開されています。この基底は `coframe()` メソッドで取り出すことが可能です。

In the above writing, the 1-form is expanded over the basis  $(dx, dy, dz)$  associated with the chart  $(x, y, z)$ . This basis can be accessed via the method `coframe()`:

```
[159]: dX = X.coframe() ; dX
```

```
[159]: ( $\mathcal{M}, (dx, dy, dz)$ )
```

多様体  $M$  の上に存在する全ての *coframe* のリストは `coframes()` メソッドを使って取り出します。

The list of all coframes defined on a given manifold open subset is returned by the method `coframes()`:

```
[160]: M.coframes()
```

```
[160]: [( $\mathcal{M}, (dx, dy, dz)$ ), ( $U, (dx, dy, dz)$ ), ( $U, (dr, d\theta, d\phi)$ )]
```

ベクトル場として、多様体上の点  $p$  における微分形式の値は `at()` メソッドで取り出します。

As for a vector field, the value of the differential form at some point on the manifold is obtained by the method `at()`:

```
[161]: dfp = df.at(p) ; print(dfp)
```

Linear form df on the Tangent space at Point p on the 3-dimensional differentiable manifold M

```
[162]: dfp.display()
```

```
[162]: df = dx + 4dy + 3dz
```

なお、 $p$  の座標は次のようなものでした。

Recall that

```
[163]: p.coord()
```

```
[163]: (1, 2, -1)
```

線形形式  $df|_p$  は  $p$  の接ベクトル空間の双対空間に所属しています。

The linear form  $df|_p$  belongs to the dual of the tangent vector space at  $p$ :

```
[164]: dfp.parent()
```

```
[164]: T_p M*
```

```
[165]: dfp.parent() is M.tangent_space(p).dual()
```

```
[165]: True
```

その結果、線形形式を  $p$  の接ベクトルに作用させると、実数が結果として得られます。

As such, it is acting on vectors at  $p$ , yielding a real number:

```
[166]: print(vp) ; vp.display()
```

Tangent vector  $v$  at Point  $p$  on the 3-dimensional differentiable manifold  $M$

```
[166]: v|_p = 3 \frac{\partial}{\partial x} - \frac{\partial}{\partial y} - 2 \frac{\partial}{\partial z}
```

```
[167]: dfp(vp)
```

```
[167]: -7
```

```
[168]: print(up) ; up.display()
```

Tangent vector  $u$  at Point  $p$  on the 3-dimensional differentiable manifold  $M$

```
[168]: u = u_x(1, 2, -1) \frac{\partial}{\partial x} + u_y(1, 2, -1) \frac{\partial}{\partial y} + u_z(1, 2, -1) \frac{\partial}{\partial z}
```

```
[169]: dfp(up)
```

```
[169]: u_x(1, 2, -1) + 4 u_y(1, 2, -1) + 3 u_z(1, 2, -1)
```

一般的な関数で定義されるスカラー場  $h$  の微分形式:

The differential 1-form of the unspecified scalar field  $h$ :

```
[170]: show(h.display())
dh = h.differential() ;
dh.display()
```

$$\begin{aligned} h: U &\longrightarrow \mathbb{R} \\ (x, y, z) &\longmapsto H(x, y, z) \\ (r, \theta, \phi) &\longmapsto H(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \end{aligned}$$

```
[170]: dh = \frac{\partial H}{\partial x} dx + \frac{\partial H}{\partial y} dy + \frac{\partial H}{\partial z} dz
```

1-形式を何もないところから、定義することもできます。

具体的な定義がまだないので、`display()` することはできません。

A 1-form can also be defined from scratch:

```
[171]: om = U.one_form(name='omega', latex_name=r'\omega')
print(om)
show(om)

try:
    om.display()
except ValueError as m:
    print("Value Error! ",m)
```

1-form  $\omega$  on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

Value Error! no basis could be found for computing the components in the Coordinate frame ( $U, (\partial / \partial x, \partial / \partial y, \partial / \partial z)$ )

与えられた coframe の要素を与えることで、この 1-形式を確定できます。

It can be specified by providing its components in a given coframe:

```
[172]: om[:] = [x^2+y^2, z, x-z] # components in the default coframe (dx,dy,dz)
om.display()
```

```
[172]:  $\omega = (x^2 + y^2) dx + zdy + (x - z) dz$ 
```

SageMath 8.8 以降では、1-形式の成分を宣言時に初期化することができます。上記の式は次の式と等価です。

Since version 8.8 of SageMath, it is possible to initialize the components of the 1-form while declaring it, so that the above is equivalent to

```
[173]: om = U.one_form(
    x^2+y^2, z, x-z, # valid only in SageMath 8.8 and higher
    name='omega',
    latex_name=r'\omega'
)
om.display()
```

```
[173]:  $\omega = (x^2 + y^2) dx + zdy + (x - z) dz$ 
```

もちろん、既定のフレームとは異なるフレームでの要素を指定することもできます。

Of course, one may set the components in a frame different from the default one:

```
[174]: om[Y.frame(), :, Y] = [r*sin(th)*cos(ph), 0, r*sin(th)*sin(ph)]
om.display(Y.frame(), Y)
```

```
[174]:  $\omega = r \cos(\phi) \sin(\theta) dr + r \sin(\phi) \sin(\theta) d\phi$ 
```

この場合、coframe  $(dx, dy, dz)$  での要素は自動的に更新されます。

The components in the coframe  $(dx, dy, dz)$  are updated automatically:

```
[175]: om.display()
```

[175]:

$$\omega = \left( \frac{x^4 + x^2 y^2 - \sqrt{x^2 + y^2 + z^2} y^2}{\sqrt{x^2 + y^2 + z^2} (x^2 + y^2)} \right) dx + \left( \frac{x^3 y + x y^3 + \sqrt{x^2 + y^2 + z^2} x y}{\sqrt{x^2 + y^2 + z^2} (x^2 + y^2)} \right) dy + \left( \frac{x z}{\sqrt{x^2 + y^2 + z^2}} \right) dz$$

元の要素の値に戻してみます。

Let us revert to the values set previously:

```
[176]: om[:] = [x^2+y^2, z, x-z]
om.display()
```

[176]:

$$\omega = (x^2 + y^2) dx + z dy + (x - z) dz$$

今回は、coframe (dr, dθ, dφ) についての要素は自動的に更新されています。

This time, the components in the coframe (dr, dθ, dφ) are those that are updated:

```
[177]: om.display(Y.frame(), Y)
```

[177]:

$$\omega = \left( r^2 \cos(\phi) \sin(\theta)^3 + r(\cos(\phi) + \sin(\phi)) \cos(\theta) \sin(\theta) - r \cos(\theta)^2 \right) dr + \left( r^2 \cos(\theta)^2 \sin(\phi) + r^2 \cos(\theta) \sin(\theta) + (r^3 \cos(\phi) \cos(\theta) - r^2 \cos(\phi)) \sin(\theta)^2 \right) d\theta + \left( -r^3 \sin(\phi) \sin(\theta)^3 + r^2 \cos(\phi) \cos(\theta) \sin(\theta) \right) d\phi$$

1-形式は、ベクトル場に作用して、スカラー場を与えます。

A 1-form acts on vector fields, resulting in a scalar field:

```
[178]: print(om(v)) ; om(v).display()
```

Scalar field omega(v) on the Open subset U of the 3-dimensional differentiable manifold M

[178]:

$$\begin{aligned} \omega(v): U &\longrightarrow \mathbb{R} \\ (x, y, z) &\longmapsto -xyz^2 + x^2y + y^3 + x^2 + y^2 + (x^2y - x)z \\ (r, \theta, \phi) &\longmapsto -r^2 \cos(\phi) \cos(\theta) \sin(\theta) + \left( r^4 \cos(\phi)^2 \cos(\theta) \sin(\phi) + r^3 \sin(\phi) \right) \sin(\theta)^3 - \left( r^4 \cos(\phi) \cos(\theta) \right) \end{aligned}$$

```
[179]: print(df(v)) ; df(v).display()
```

Scalar field df(v) on the Open subset U of the 3-dimensional differentiable manifold M

[179]:

$$\begin{aligned} df(v): U &\longrightarrow \mathbb{R} \\ (x, y, z) &\longmapsto 3xyz^3 - (2x - 1)y + 1 \\ (r, \theta, \phi) &\longmapsto r \sin(\phi) \sin(\theta) + \left( 3r^5 \cos(\phi) \cos(\theta)^3 \sin(\phi) - 2r^2 \cos(\phi) \sin(\phi) \right) \sin(\theta)^2 + 1 \end{aligned}$$

```
[180]: om(u).display()
```

[180]:  $\omega(u): U \rightarrow \mathbb{R}$   
 $(x, y, z) \mapsto x^2 u_x(x, y, z) + y^2 u_x(x, y, z) + z(u_y(x, y, z) - u_z(x, y, z)) + x u_z(x, y, z)$   
 $(r, \theta, \phi) \mapsto r^2 \sin(\theta)^2 u_x(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) + r \cos(\theta) u_y(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta))$

スカラー場  $f$  から導かれた 1-形式については、次の恒等式が成立します。

In the case of a differential 1-form, the following identity holds:

[181]: `df(v) == v(f)`

[181]: True

1-形式は、親が  $U$  上で定義された全ての 1-形式からなる  $C^\infty(U)$  級のモジュール  $\Omega^1(U)$  である Sage の *element* オブジェクトです。

1-forms are Sage *element* objects, whose *parent* is the  $C^\infty(U)$ -module  $\Omega^1(U)$  of all 1-forms defined on  $U$ :

[182]: `df.parent()`

[182]:  $\Omega^1(U)$

[183]: `print(df.parent())`

Free module  $\Omega^1(U)$  of 1-forms on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

[184]: `print(om.parent())`

Free module  $\Omega^1(U)$  of 1-forms on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

$\Omega^1(U)$  は自由モジュール  $\mathfrak{X}(U)$  の双対です。

$\Omega^1(U)$  is actually the dual of the free module  $\mathfrak{X}(U)$ :

[185]: `df.parent() is v.parent().dual()`

[185]: True

## 1.8 微分形式と外積, 外微分 :Differential forms and exterior calculus

二つの 1-形式の外積は `wedge()` メソッドを使って求められます。その結果は 2-形式となります。

Differential forms and exterior calculus

The exterior product of two 1-forms is taken via the method `wedge()` and results in a 2-form:

[186]: `a = om.wedge(df) ; print(a) ; a.display()`

2-form  $\omega \wedge df$  on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

[186]:  $\omega \wedge df = (2x^2y + 2y^3 - z) dx \wedge dy + (3(x^2 + y^2)z^2 - x + z) dx \wedge dz + (3z^3 - 2xy + 2yz) dy \wedge dz$   
 行列形式での要素の表示:

A matrix view of the components:

[187]: `a[:]`

[187]: 
$$\begin{pmatrix} 0 & 2x^2y + 2y^3 - z & 3(x^2 + y^2)z^2 - x + z \\ -2x^2y - 2y^3 + z & 0 & 3z^3 - 2xy + 2yz \\ -3(x^2 + y^2)z^2 + x - z & -3z^3 + 2xy - 2yz & 0 \end{pmatrix}$$

歪対称性から導かれる余分な要素を省き、ゼロでない成分だけを表示します:

Displaying only the non-vanishing components, skipping the redundant ones (i.e. those that can be deduced by antisymmetry):

[188]: `a.display_comp(only_nonredundant=True)`

[188]: 
$$\begin{aligned} \omega \wedge df_{xy} &= 2x^2y + 2y^3 - z \\ \omega \wedge df_{xz} &= 3(x^2 + y^2)z^2 - x + z \\ \omega \wedge df_{yz} &= 3z^3 - 2xy + 2yz \end{aligned}$$

2-形式  $\omega \wedge df$  を コフレーム  $(dr, d\theta, d\phi)$  での成分に展開することもできます。

The 2-form  $\omega \wedge df$  can be expanded on the  $(dr, d\theta, d\phi)$  coframe:

[189]: `a.display(Y.frame(), Y)`

[189]: 
$$\begin{aligned} \omega \wedge df &= \left(3r^5 \cos(\phi) \sin(\theta)^4 - \left(3r^5 \cos(\phi) - 3r^4 \cos(\theta) \sin(\phi) - 2r^3 \cos(\phi) \sin(\phi)^2\right) \sin(\theta)^2 - (3r^4 \sin(\phi) + r^2 \cos(\phi)) \right. \\ & d\theta + \left(2r^4 \sin(\phi) \sin(\theta)^5 + \left(3r^5 \cos(\theta)^3 \sin(\phi) + 2r^3 \cos(\phi)^2 \cos(\theta) \sin(\phi)\right) \sin(\theta)^3 - \left(2r^3 \cos(\phi) \cos(\theta)^2 \sin(\phi) + \right. \\ & d\phi + \left(-r^3 \cos(\theta)^2 \sin(\theta) - \left(3r^6 \cos(\theta)^2 \sin(\phi) + 2r^4 \cos(\phi)^2 \sin(\phi) - 2r^5 \cos(\theta) \sin(\phi)\right) \sin(\theta)^4 + (2r^4 \cos(\phi) \cos(\theta) \right. \\ & \left. \left. d\phi \right) \end{aligned}$$

$A := \omega \wedge df$  は 2-形式として二組のベクトルに作用させることができ、また歪対称です。

As a 2-form,  $A := \omega \wedge df$  can be applied to a pair of vectors and is antisymmetric:

[190]: `a.set_name('A')`  
`print(a(u,v)) ; a(u,v).display()`

Scalar field  $A(u,v)$  on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

[190]: 
$$\begin{aligned} A(u,v) : U &\longrightarrow \mathbb{R} \\ (x,y,z) &\longmapsto 3xyz^4u_y(x,y,z) - 2x^2y^2u_y(x,y,z) - 2y^4u_y(x,y,z) - 2(xu_x(x,y,z) + u_y(x,y,z))y^3 - \\ (r,\theta,\phi) &\longmapsto \left(r^4 \cos(\phi) \cos(\theta)^2 \sin(\phi) \sin(\theta)^2 + \left(\sin(\phi)^3 - \sin(\phi)\right)r^4 \cos(\theta) \sin(\theta)^3 + r^2 \cos(\phi) \cos(\theta) \right) \end{aligned}$$

[191]: `a(u,v) == - a(v,u)`

[191]: True



```
[192]: a.symmetries()
```

```
no symmetry; antisymmetry: (0, 1)
```

ある微分形式の外微分は以下のように、`exterior_derivative()` メソッドを使って求めます。

The exterior derivative of a differential form:

```
[193]: dom = om.exterior_derivative() ; print(dom) ; dom.display()
```

```
2-form domega on the Open subset U of the 3-dimensional differentiable manifold
M
```

```
[193]: dω = -2ydx ∧ dy + dx ∧ dz - dy ∧ dz
```

`exterior_derivative()` メソッドを作用させる代わりに、関数 `diff()` を使うこともできます (SageMath 9.2 以降で利用可能)。

Instead of invoking the method `exterior_derivative()`, one can use the function `diff()` (available in SageMath 9.2 or higher):

```
[194]: dom = diff(om)
dom
```

```
[194]: dω
```

```
[195]: da = diff(a) ; print(da) ; da.display()
```

```
3-form dA on the Open subset U of the 3-dimensional differentiable manifold M
```

```
[195]: dA = (-6yz2 - 2y - 1) dx ∧ dy ∧ dz
```

外微分を続けて作用させると、結果は0になります。

The exterior derivative is nilpotent:

```
[196]: ddf = diff(df) ; ddf.display()
```

```
[196]:ddf = 0
```

```
[197]: ddom = diff(dom) ; ddom.display()
```

```
[197]:ddω = 0
```

## 1.9 Lie 微分: Lie derivative

テンソル場のあるベクトル場に関する Lie 微分は、`lie_derivative()` メソッドを用いて計算します。

The Lie derivative of any tensor field with respect to a vector field is computed by the method `lie_derivative()`, with the vector field as the argument:

```
[198]: lv_om = om.lie_derivative(v) ; print(lv_om) ;
lv_om.display()
```

1-form on the Open subset U of the 3-dimensional differentiable manifold M

[198]:  $(-yz^2 + (xy - 1)z + 2x) dx + (-xz^2 + x^2 + y^2 + (x^2 + xy)z) dy + (-2xyz + (x^2 + 1)y + 1) dz$

```
[199]: lu_dh = dh.lie_derivative(u) ; print(lu_dh) ;
lu_dh.display()
```

1-form on the Open subset U of the 3-dimensional differentiable manifold M

[199]: 
$$\left( u_x(x, y, z) \frac{\partial^2 H}{\partial x^2} + u_y(x, y, z) \frac{\partial^2 H}{\partial x \partial y} + u_z(x, y, z) \frac{\partial^2 H}{\partial x \partial z} + \frac{\partial H}{\partial x} \frac{\partial u_x}{\partial x} + \frac{\partial H}{\partial y} \frac{\partial u_y}{\partial x} + \frac{\partial H}{\partial z} \frac{\partial u_z}{\partial x} \right) dx +$$

$$\left( u_x(x, y, z) \frac{\partial^2 H}{\partial x \partial y} + u_y(x, y, z) \frac{\partial^2 H}{\partial y^2} + u_z(x, y, z) \frac{\partial^2 H}{\partial y \partial z} + \frac{\partial H}{\partial x} \frac{\partial u_x}{\partial y} + \frac{\partial H}{\partial y} \frac{\partial u_y}{\partial y} + \frac{\partial H}{\partial z} \frac{\partial u_z}{\partial y} \right) dy +$$

$$\left( u_x(x, y, z) \frac{\partial^2 H}{\partial x \partial z} + u_y(x, y, z) \frac{\partial^2 H}{\partial y \partial z} + u_z(x, y, z) \frac{\partial^2 H}{\partial z^2} + \frac{\partial H}{\partial x} \frac{\partial u_x}{\partial z} + \frac{\partial H}{\partial y} \frac{\partial u_y}{\partial z} + \frac{\partial H}{\partial z} \frac{\partial u_z}{\partial z} \right) dz$$

Cartan の恒等式を確かめてみましょう

1-形式  $\omega$  についての Cartan の恒等式

$$\mathcal{L}_v \omega = v \cdot d\omega + d\langle \omega, v \rangle$$

2-形式  $\omega$  についての Cartan の恒等式

$$\mathcal{L}_v A = v \cdot dA + d(v \cdot A)$$

Let us check Cartan identity on the 1-form  $\omega$ :

$$\mathcal{L}_v \omega = v \cdot d\omega + d\langle \omega, v \rangle$$

and on the 2-form A:

$$\mathcal{L}_v A = v \cdot dA + d(v \cdot A)$$

```
[200]: om.lie_derivative(v) == v.contract(diff(om)) + diff(om(v))
```

[200]: True

```
[201]: a.lie_derivative(v) == v.contract(diff(a)) + diff(v.contract(a))
```

[201]: True

ベクトル場の別のベクトル場による Lie 微分は、二つのベクトル場による交換子積 (あるいは 交換子) になります。

The Lie derivative of a vector field along another one is the commutator of the two vectors fields:

```
[202]: v.lie_derivative(u)(f) == u(v(f)) - v(u(f))
```

[202]: True

## 1.10 任意の階数のテンソル場:Tensor fields of arbitrary rank

これまでに、次のようなテンソル場をみてきました。

- (0,0) 形式のテンソル : スカラー場,
- (1,0) 形式のテンソル : ベクトル場,
- (0,1) 形式のテンソル : 1-形式あるいは一階の微分形式,
- (0,2) 形式で歪対称のテンソル : 2-形式あるいは二階の微分形式.

より一般に、 $(p, q)$  型のテンソル場を SageManifolds に導入することができます。例えば、副多様体  $U$  上の (1,2) 型のテンソル場を次のように宣言します。

Up to now, we have encountered tensor fields

- of type (0,0) (i.e. scalar fields),
- of type (1,0) (i.e. vector fields),
- of type (0,1) (i.e. 1-forms),
- of type (0,2) and antisymmetric (i.e. 2-forms).

More generally, tensor fields of any type  $(p, q)$  can be introduced in SageMath. For instance a tensor field of type (1,2) on the open subset  $U$  is declared as follows:

```
[203]: t = U.tensor_field(1, 2, name='T') ; print(t)
```

Tensor field T of type (1,2) on the Open subset U of the 3-dimensional differentiable manifold M

ベクトルや 1-形式と同じように、テンソル場の既定のフレームに関する要素を [] 演算子を使って設定できます。

As for vectors or 1-forms, the tensor's components with respect to the domain's default frame are set by means of square brackets:

```
[204]: t[1,2,1] = 1 + x^2
t[3,2,1] = x*y*z
```

設定されていない要素は 0 です。

Unset components are zero:

```
[205]: t.display()
```

```
[205]: T = (x^2 + 1) \frac{\partial}{\partial x} \otimes dy \otimes dx + xyz \frac{\partial}{\partial z} \otimes dy \otimes dx
```

```
[206]: t[:]
```

```
[206]: [[ [0, 0, 0], [x^2 + 1, 0, 0], [0, 0, 0] ], [ [0, 0, 0], [0, 0, 0], [0, 0, 0] ], [ [0, 0, 0], [xyz, 0, 0], [0, 0, 0] ] ]
```

0 でない要素を表示します。

Display of the nonzero components:

```
[207]: t.display_comp()
```

[207]: 
$$\begin{aligned} T^x_{yx} &= x^2 + 1 \\ T^z_{yx} &= xyz \end{aligned}$$

二重括弧 [[]] 演算子は既定のフレームに関する成分をスカラー場として返します。一方、一重の括弧 [] 演算子はこのスカラー場の既定の座標系についての表現を返します。

Double square brackets return the component (still w.r.t. the default frame) as a scalar field, while single square brackets return the expression of this scalar field in terms of the domain's default coordinates:

[208]: 

```
print(t[[1,2,1]]) ;
t[[1,2,1]].display()
```

Scalar field on the Open subset U of the 3-dimensional differentiable manifold M

[208]: 
$$\begin{aligned} U &\longrightarrow \mathbb{R} \\ (x, y, z) &\longmapsto x^2 + 1 \\ (r, \theta, \phi) &\longmapsto r^2 \cos(\phi)^2 \sin(\theta)^2 + 1 \end{aligned}$$

[209]: 

```
print(t[1,2,1])
t[1,2,1]
```

$x^2 + 1$

[209]:  $x^2 + 1$

(1,2) 型のテンソル場は (1-形式、ベクトル場、ベクトル場) からなる 3 成分のタプルをスカラー場に写像します。

A tensor field of type (1,2) maps a 3-tuple (1-form, vector field, vector field) to a scalar field:

[210]: 

```
print(t(om, u, v))
t(om, u, v).display()
```

Scalar field  $T(\omega, u, v)$  on the Open subset U of the 3-dimensional differentiable manifold M

[210]: 
$$\begin{aligned} T(\omega, u, v): U &\longrightarrow \mathbb{R} \\ (x, y, z) &\longmapsto (x^2 + 1)y^3u_y(x, y, z) + (x^2 + 1)y^2u_y(x, y, z) - (xy^2u_y(x, y, z) + xyu_y(x, y, z))z^2 + \\ (r, \theta, \phi) &\longmapsto \left( r^5 \cos(\phi)^2 \sin(\phi) \sin(\theta)^5 - \left( (\cos(\phi)^4 - \cos(\phi)^2) r^5 \cos(\theta) - r^4 \cos(\phi)^2 \right) \sin(\theta)^4 + \right. \end{aligned}$$

ベクトル場や微分形式と同じように、多様体に定義されたどのフレームについてのテンソル場の成分も取り出せます。

As for vectors and differential forms, the tensor components can be taken in any frame defined on the manifold:

[211]: 

```
t[Y.frame(), 1,1,1, Y]
```

[211]: 
$$r^2 \cos(\phi)^4 \sin(\phi) \sin(\theta)^5 + (\cos(\phi)^4 - \cos(\phi)^2) r^3 \sin(\theta)^6 - (\cos(\phi)^4 - \cos(\phi)^2) r^3 \sin(\theta)^4 + \cos(\phi)^2 \sin(\phi) \sin(\theta)^3$$

## 1.11 テンソル計算 : Tensor calculus

テンソル積  $\otimes$  は \* で示されます。

The tensor product  $\otimes$  is denoted by \*:

```
[212]: print(v.tensor_type())
       print(a.tensor_type())
```

```
(1, 0)
(0, 2)
```

```
[213]: b = v*a
       print(b)
       b
```

Tensor field  $v \otimes A$  of type (1,2) on the Open subset U of the 3-dimensional differentiable manifold M

```
[213]: v \otimes A
```

テンソル積は、(歪)対称性を保ちます。A は 2-形式なので、二つの引数 (位置 0 および 1) について歪対称です。結果として b は最後の二つの引数 (位置 1 および 2) について歪対称となります。

The tensor product preserves the (anti)symmetries: since A is a 2-form, it is antisymmetric with respect to its two arguments (positions 0 and 1); as a result, b is antisymmetric with respect to its last two arguments (positions 1 and 2):

```
[214]: a.symmetries()
```

```
no symmetry; antisymmetry: (0, 1)
```

```
[215]: b.symmetries()
```

```
no symmetry; antisymmetry: (1, 2)
```

標準的な テンソル 計算 は実装済みです。

Standard tensor arithmetics is implemented:

```
[216]: s = - t + 2*f* b
       print(s)
```

Tensor field of type (1,2) on the Open subset U of the 3-dimensional differentiable manifold M

テンソルの縮約は `trace()` メソッドおよび `contract()` メソッドで取り扱われます。例として、テンソル  $T$  の最初の二つの引数 (位置 0 および 1) について縮約してみましょう。つまり、 $c_i = T^k_{ki}$  で与えられるテンソル  $c$  を構築してみましょう。

**Tensor contractions** are dealt with by the method `trace()` and `contract()`: for instance, let us contract the tensor  $T$  w.r.t. its first two arguments (positions 0 and 1), i.e. let us form the tensor  $c$  of components  $c_i = T^k_{ki}$ :

```
[217]: c = t.trace(0,1)
print(c)
```

1-form on the Open subset U of the 3-dimensional differentiable manifold M

`trace(0,1)` と入力する代わりに、指標表記を使って縮約を指定できます。指標は文字列として [] 演算子の中に記述されます。文字列中で、“^” は反変指標の前、“\_” は共変指標の前、に置かれます。

An alternative to the writing `trace(0,1)` is to use the **index notation** to denote the contraction: the indices are given in a string inside the [] operator, with '^' in front of the contravariant indices and '\_' in front of the covariant ones:

```
[218]: c1 = t['^k_ki']
print(c1)
c1 == c
```

1-form on the Open subset U of the 3-dimensional differentiable manifold M

```
[218]: True
```

縮約は繰り返された指標 (ここでは  $k$  です) について実行されます。それ以外の文字 (ここでは  $i$ ) は任意です。

The contraction is performed on the repeated index (here  $k$ ); the letter denoting the remaining index (here  $i$ ) is arbitrary:

```
[219]: t['^k_kj'] == c
```

```
[219]: True
```

```
[220]: t['^b_ba'] == c
```

```
[220]: True
```

縮約されない指標については “.” (ドット) で置き換え可能です。

It can even be replaced by a dot:

```
[221]: t['^k_k.'] == c
```

```
[221]: True
```

この文字列中では LaTeX 式の指定も可能です。

LaTeX notations are allowed:

```
[222]: t['^{k}_{ki}'] == c
```

```
[222]: True
```

SageMath 9.2 以降では、ギリシャ文字を使うことも可能です。

as well as Greek letters (only for SageMath 9.2 or higher):

```
[223]: t['^{\mu}_{\mu\alpha}'] == c
```

[223]: True

テンソル場  $T$  とベクトル場  $v$  の縮約  $T^i_{jk}v^k$  は次のようにして、実行されます。ここで、2 は  $T$  の最後の指標の場所、0 は  $v$  の唯一の指標の場所です。

The contraction  $T^i_{jk}v^k$  of the tensor fields  $T$  and  $v$  is taken as follows (2 refers to the last index position of  $T$  and 0 to the only index position of  $v$ ):

```
[224]: tv = t.contract(2, v, 0)
print(tv)
```

Tensor field of type (1,1) on the Open subset U of the 3-dimensional differentiable manifold M

2 は  $T$  の最後の指標の場所、0 は  $v$  の唯一の指標の場所ですので、上記の表記の簡略版は次のようになります。

Since 2 corresponds to the last index position of  $T$  and 0 to the first index position of  $v$ , a shortcut for the above is

```
[225]: tv1 = t.contract(v)
print(tv1)
```

Tensor field of type (1,1) on the Open subset U of the 3-dimensional differentiable manifold M

```
[226]: tv1 == tv
```

[226]: True

`contract()` メソッドに替わって、指標表記と直積演算子`*`を組み合わせることで、縮約を表現できます。

Instead of `contract()`, the index notation, combined with the `*` operator, can be used to denote the contraction:

```
[227]: t['^i_jk']*v['^k'] == tv
```

[227]: True

繰り返し替えされない指標については、`.`で置き換えることができます。

The non-repeated indices can be replaced by dots:

```
[228]: t['^._.k']*v['^k'] == tv
```

[228]: True

## 1.12 計量構造: Metric structures

多様体  $\mathcal{M}$  の Riemann 計量はつぎのように宣言されます。

A Riemannian metric on the manifold  $\mathcal{M}$  is declared as follows:

```
[229]: g = M.riemannian_metric('g')
print(g)
```

Riemannian metric  $g$  on the 3-dimensional differentiable manifold  $M$

計量  $g$  は対称な (0,2) 型のテンソル場です。

It is a symmetric tensor field of type (0,2):

```
[230]: g.parent()
```

```
[230]:  $\mathcal{T}^{(0,2)}(\mathcal{M})$ 
```

```
[231]: print(g.parent())
```

Free module  $\mathcal{T}^{(0,2)}(M)$  of type-(0,2) tensors fields on the 3-dimensional differentiable manifold  $M$

```
[232]: g.symmetries()
```

symmetry: (0, 1); no antisymmetry

計量は、あるベクトルフレームに関する要素で初期化されます。例えば、 $\mathcal{M}$  の既定のフレームを使って、

The metric is initialized by its components with respect to some vector frame. For instance, using the default frame of  $\mathcal{M}$ :

```
[233]: g[1,1], g[2,2], g[3,3] = 1, 1, 1
g.display()
```

```
[233]:  $g = dx \otimes dx + dy \otimes dy + dz \otimes dz$ 
```

別のベクトルフレームに関する成分を表示します。

The components w.r.t. another vector frame are obtained as for any tensor field:

```
[234]: g.display(Y.frame(), Y)
```

```
[234]:  $g = dr \otimes dr + r^2 d\theta \otimes d\theta + r^2 \sin(\theta)^2 d\phi \otimes d\phi$ 
```

```
[235]: show(u.display())
```

$$u = u_x(x, y, z) \frac{\partial}{\partial x} + u_y(x, y, z) \frac{\partial}{\partial y} + u_z(x, y, z) \frac{\partial}{\partial z}$$

もちろん計量は二つのベクトルに作用します。

Of course, the metric acts on vector pairs:

```
[236]: show(u.display()); show(v.display());
print(g(u,v)) ;
g(u,v).display()
```



$$u = u_x(x, y, z) \frac{\partial}{\partial x} + u_y(x, y, z) \frac{\partial}{\partial y} + u_z(x, y, z) \frac{\partial}{\partial z}$$

$$v = (y+1) \frac{\partial}{\partial x} - x \frac{\partial}{\partial y} + xyz \frac{\partial}{\partial z}$$

Scalar field  $g(u,v)$  on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

[236]:  $g(u, v) : U \rightarrow \mathbb{R}$

$$(x, y, z) \mapsto xyz u_z(x, y, z) + y u_x(x, y, z) - x u_y(x, y, z) + u_x(x, y, z)$$

$$(r, \theta, \phi) \mapsto r^3 \cos(\phi) \cos(\theta) \sin(\phi) \sin(\theta)^2 u_z(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) - r \cos(\phi) \sin(\theta)$$

計量  $g$  に付随する **Levi-Civita** 接続は、`.connection()` メソッドを使って求められます。 :

The Levi-Civita connection associated to the metric  $g$ :

[237]: 

```
nabla = g.connection()
print(nabla)
nabla
```

Levi-Civita connection `nabla_g` associated with the Riemannian metric  $g$  on the 3-dimensional differentiable manifold  $M$

[237]:  $\nabla_g$

多様体の既定の座標をつかって Christoffel 記号を書き出します。既定の座標系では、すべての成分が 0 になっています。

The Christoffel symbols with respect to the manifold's default coordinates:

[238]: 

```
nabla.coef()[:]
```

[238]:  $[[[0, 0, 0], [0, 0, 0], [0, 0, 0]], [[0, 0, 0], [0, 0, 0], [0, 0, 0]], [[0, 0, 0], [0, 0, 0], [0, 0, 0]]]$

次に、球面座標  $(r, \theta, \phi)$  を使って、Christoffel 記号を書き出します。空間の各点で、接ベクトルの向きが変わっていくことから、接続係数にも非零の成分が現れます。

The Christoffel symbols with respect to the coordinates  $(r, \theta, \phi)$ :

[239]: 

```
nabla.coef(Y.frame())[:, Y]
```

[239]:  $\left[ \left[ [0, 0, 0], [0, -r, 0], [0, 0, -r \sin(\theta)^2] \right], \left[ \left[ 0, \frac{1}{r}, 0 \right], \left[ \frac{1}{r}, 0, 0 \right], [0, 0, -\cos(\theta) \sin(\theta)] \right], \left[ \left[ 0, 0, \frac{1}{r} \right], \left[ 0, 0, \frac{\cos(\theta)}{\sin(\theta)} \right], \left[ \frac{1}{r}, 0, 0 \right] \right] \right]$

`display()` を使って、見易く表示させて見ます。既定の設定では非零の接続係数のみが表示されます。

A nice view is obtained via the method `display()` (by default, only the nonzero connection coefficients are shown):

[240]: 

```
nabla.display(frame=Y.frame(), chart=Y)
```

[240]:

$$\begin{aligned}
\Gamma^r_{\theta\theta} &= -r \\
\Gamma^r_{\phi\phi} &= -r \sin(\theta)^2 \\
\Gamma^\theta_{r\theta} &= \frac{1}{r} \\
\Gamma^\theta_{\theta r} &= \frac{1}{r} \\
\Gamma^\theta_{\phi\phi} &= -\cos(\theta) \sin(\theta) \\
\Gamma^\phi_{r\phi} &= \frac{1}{r} \\
\Gamma^\phi_{\theta\phi} &= \frac{\cos(\theta)}{\sin(\theta)} \\
\Gamma^\phi_{\phi r} &= \frac{1}{r} \\
\Gamma^\phi_{\phi\theta} &= \frac{\cos(\theta)}{\sin(\theta)}
\end{aligned}$$

計量の `christoffel_symbols_display()` メソッドを使うこともできます。この関数は（既定の設定では）独立な Christoffel 記号だけを表示します。

One may also use the method `christoffel_symbols_display()` of the metric, which (by default) displays only the non-redundant Christoffel symbols:

```
[241]: g.christoffel_symbols_display(Y)
```

```
[241]:
```

$$\begin{aligned}
\Gamma^r_{\theta\theta} &= -r \\
\Gamma^r_{\phi\phi} &= -r \sin(\theta)^2 \\
\Gamma^\theta_{r\theta} &= \frac{1}{r} \\
\Gamma^\theta_{\phi\phi} &= -\cos(\theta) \sin(\theta) \\
\Gamma^\phi_{r\phi} &= \frac{1}{r} \\
\Gamma^\phi_{\theta\phi} &= \frac{\cos(\theta)}{\sin(\theta)}
\end{aligned}$$

### 1.12.1 接続の共変微分としての作用。

The connection acting as a covariant derivative:

```
[242]: nab_v = nabla(v)
print(nab_v) ; nab_v.display()
```

Tensor field `nabla_g(v)` of type (1,1) on the Open subset U of the 3-dimensional differentiable manifold M

```
[242]:
```

$$\nabla_g v = \frac{\partial}{\partial x} \otimes dy - \frac{\partial}{\partial y} \otimes dx + yz \frac{\partial}{\partial z} \otimes dx + xz \frac{\partial}{\partial z} \otimes dy + xy \frac{\partial}{\partial z} \otimes dz$$

```
[243]: v.display(Y)
```

```
[243]:
```

$$v = \left( r^3 \cos(\phi) \cos(\theta)^2 \sin(\phi) \sin(\theta)^2 + \cos(\phi) \sin(\theta) \right) \frac{\partial}{\partial r} + \left( -\frac{r^3 \cos(\phi) \cos(\theta) \sin(\phi) \sin(\theta)^3 - \cos(\phi) \cos(\theta)}{r} \right) \frac{\partial}{\partial \theta} + \left( -\frac{r \sin(\theta) + \sin(\phi)}{r \sin(\theta)} \right) \frac{\partial}{\partial \phi}$$

```
[244]: v.display()
```

```
[244]:
```

$$v = (y + 1) \frac{\partial}{\partial x} - x \frac{\partial}{\partial y} + xyz \frac{\partial}{\partial z}$$

```
[245]: nab_v.display(Y)
```

$$\begin{aligned}
[245]: \quad \nabla_g v = & 3r^2 \cos(\phi) \cos(\theta)^2 \sin(\phi) \sin(\theta)^2 \frac{\partial}{\partial r} \otimes dr + \left( 3r^3 \cos(\phi) \cos(\theta)^3 \sin(\phi) - r^3 \cos(\phi) \cos(\theta) \sin(\phi) \right) \sin(\theta) \frac{\partial}{\partial r} \otimes \\
& d\theta + \left( \left( 2 \cos(\phi)^2 - 1 \right) r^3 \cos(\theta)^2 + r \right) \sin(\theta)^2 \frac{\partial}{\partial r} \otimes d\phi - 3r \cos(\phi) \cos(\theta) \sin(\phi) \sin(\theta)^3 \frac{\partial}{\partial \theta} \otimes \\
& dr + \left( 3r^2 \cos(\phi) \cos(\theta)^4 \sin(\phi) - 4r^2 \cos(\phi) \cos(\theta)^2 \sin(\phi) + r^2 \cos(\phi) \sin(\phi) \right) \frac{\partial}{\partial \theta} \otimes d\theta + \\
& \left( \left( 2 \sin(\phi)^2 - 1 \right) r^2 \cos(\theta) \sin(\theta)^3 + \cos(\theta) \sin(\theta) \right) \frac{\partial}{\partial \theta} \otimes d\phi - \frac{1}{r} \frac{\partial}{\partial \phi} \otimes dr - \frac{\cos(\theta)}{\sin(\theta)} \frac{\partial}{\partial \phi} \otimes d\theta
\end{aligned}$$

```
[246]: nab_v.display()
```

$$[246]: \quad \nabla_g v = \frac{\partial}{\partial x} \otimes dy - \frac{\partial}{\partial y} \otimes dx + yz \frac{\partial}{\partial z} \otimes dx + xz \frac{\partial}{\partial z} \otimes dy + xy \frac{\partial}{\partial z} \otimes dz$$

Levi-Civita 接続係数であることから、 $\nabla_g$  には捩れ (torsion) はありません。

Being a Levi-Civita connection,  $\nabla_g$  is torsion.free:

Levi-Civita 接続は定義から、対称な接続係数なので、捩れはない。

```
[247]: print(nabla.torsion()); nabla.torsion().display()
```

Tensor field of type (1,2) on the 3-dimensional differentiable manifold M

```
[247]: 0
```

この例ではさらに、多様体は平坦 (フラット, 曲率が0) です。

In the present case, it is also flat:

```
[248]: print(nabla.riemann()); nabla.riemann().display()
```

Tensor field Riem(g) of type (1,3) on the 3-dimensional differentiable manifold M

```
[248]: Riem(g) = 0
```

### 1.12.2 曲率のある空間

$g_{rr}$  を  $1/(1+r^2)$  と変更することで、曲率のある空間を考えましょう。

Let us consider a non-flat metric, by changing  $g_{rr}$  to  $1/(1+r^2)$ :

```
[249]: g[Y.frame(), 1,1, Y] = 1/(1+r^2)
g.display(Y.frame(), Y)
```

$$[249]: \quad g = \left( \frac{1}{r^2 + 1} \right) dr \otimes dr + r^2 d\theta \otimes d\theta + r^2 \sin(\theta)^2 d\phi \otimes d\phi$$

簡単の為に、部分多様体  $U$  の既定の座標系を  $Y=(U, (r, \theta, \phi))$  に変更します。

For convenience, we change the default chart on the domain  $U$  to  $Y=(U, (r, \theta, \phi))$ :

```
[250]: U.set_default_chart(Y)
```

これによって、 $(r, \theta, \phi)$  を使った座標表現を求める際に座標系  $Y$  を指定する必要がありません。

In this way, we do not have to specify  $Y$  when asking for coordinate expressions in terms of  $(r, \theta, \phi)$ :

```
[251]: g.display(Y.frame())
```

```
[251]:
```

$$g = \left( \frac{1}{r^2 + 1} \right) dr \otimes dr + r^2 d\theta \otimes d\theta + r^2 \sin(\theta)^2 d\phi \otimes d\phi$$

この計量は双曲空間  $\mathbb{H}^3$  の計量だと気がつきます。座標系  $(U, (x, y, z))$  についての表現は、次のようになります。

We recognize the metric of the hyperbolic space  $\mathbb{H}^3$ . Its expression in terms of the chart  $(U, (x, y, z))$  is

```
[252]: g.display(X_U.frame(), X_U)
```

```
[252]:
```

$$g = \left( \frac{y^2 + z^2 + 1}{x^2 + y^2 + z^2 + 1} \right) dx \otimes dx + \left( -\frac{xy}{x^2 + y^2 + z^2 + 1} \right) dx \otimes dy + \left( -\frac{xz}{x^2 + y^2 + z^2 + 1} \right) dx \otimes dz + \left( -\frac{xy}{x^2 + y^2 + z^2 + 1} \right) dy \otimes dx + \left( \frac{x^2 + z^2 + 1}{x^2 + y^2 + z^2 + 1} \right) dy \otimes dy + \left( -\frac{yz}{x^2 + y^2 + z^2 + 1} \right) dy \otimes dz + \left( -\frac{xz}{x^2 + y^2 + z^2 + 1} \right) dz \otimes dx + \left( -\frac{yz}{x^2 + y^2 + z^2 + 1} \right) dz \otimes dy + \left( \frac{x^2 + y^2 + 1}{x^2 + y^2 + z^2 + 1} \right) dz \otimes dz$$

要素の行列表示の方がより適切かもしれません。

A matrix view of the components may be more appropriate:

```
[253]: g[X_U.frame(), :, X_U]
```

```
[253]:
```

$$\begin{pmatrix} \frac{y^2 + z^2 + 1}{x^2 + y^2 + z^2 + 1} & -\frac{xy}{x^2 + y^2 + z^2 + 1} & -\frac{xz}{x^2 + y^2 + z^2 + 1} \\ -\frac{xy}{x^2 + y^2 + z^2 + 1} & \frac{x^2 + z^2 + 1}{x^2 + y^2 + z^2 + 1} & -\frac{yz}{x^2 + y^2 + z^2 + 1} \\ -\frac{xz}{x^2 + y^2 + z^2 + 1} & -\frac{yz}{x^2 + y^2 + z^2 + 1} & \frac{x^2 + y^2 + 1}{x^2 + y^2 + z^2 + 1} \end{pmatrix}$$

もともとは  $U$  でのみ定義されていたこれらの成分を、全体の多様体  $\mathcal{M}$  までに拡張します。拡張は、座標系  $X=(\mathcal{M}, (x, y, z))$  に付随するフレームでの座標表現が同じであることを要求して実行されます。

We extend these components, a priori defined only on  $U$ , to the whole manifold  $\mathcal{M}$ , by demanding the same coordinate expressions in the frame associated to the chart  $X=(\mathcal{M}, (x, y, z))$ :

```
[254]: g.add_comp_by_continuation(X.frame(), U, X)
g.display()
```

```
[254]:
```

$$g = \left( \frac{y^2 + z^2 + 1}{x^2 + y^2 + z^2 + 1} \right) dx \otimes dx + \left( -\frac{xy}{x^2 + y^2 + z^2 + 1} \right) dx \otimes dy + \left( -\frac{xz}{x^2 + y^2 + z^2 + 1} \right) dx \otimes dz + \left( -\frac{xy}{x^2 + y^2 + z^2 + 1} \right) dy \otimes dx + \left( \frac{x^2 + z^2 + 1}{x^2 + y^2 + z^2 + 1} \right) dy \otimes dy + \left( -\frac{yz}{x^2 + y^2 + z^2 + 1} \right) dy \otimes dz + \left( -\frac{xz}{x^2 + y^2 + z^2 + 1} \right) dz \otimes dx + \left( -\frac{yz}{x^2 + y^2 + z^2 + 1} \right) dz \otimes dy + \left( \frac{x^2 + y^2 + 1}{x^2 + y^2 + z^2 + 1} \right) dz \otimes dz$$

Levi-Civita 接続は、計量  $g$  の変更後、自動的に再計算されます。

The Levi-Civita connection is automatically recomputed, after the change in  $g$ :

```
[255]: nabla = g.connection()
```

とくに、Christoffel 係数は、異なっています。

In particular, the Christoffel symbols are different:

```
[256]: nabla.display(only_nonredundant=True)
```

```
[256]:
```

$$\begin{aligned} \Gamma^x_{xx} &= -\frac{xy^2+xz^2+x}{x^2+y^2+z^2+1} \\ \Gamma^x_{xy} &= \frac{x^2y}{x^2+y^2+z^2+1} \\ \Gamma^x_{xz} &= \frac{x^2z}{x^2+y^2+z^2+1} \\ \Gamma^x_{yy} &= -\frac{x^3+xz^2+x}{x^2+y^2+z^2+1} \\ \Gamma^x_{yz} &= \frac{xyz}{x^2+y^2+z^2+1} \\ \Gamma^x_{zz} &= -\frac{x^3+xy^2+x}{x^2+y^2+z^2+1} \\ \Gamma^y_{xx} &= -\frac{y^3+yz^2+y}{x^2+y^2+z^2+1} \\ \Gamma^y_{xy} &= \frac{xy^2}{x^2+y^2+z^2+1} \\ \Gamma^y_{xz} &= \frac{xyz}{x^2+y^2+z^2+1} \\ \Gamma^y_{yy} &= -\frac{yz^2+(x^2+1)y}{x^2+y^2+z^2+1} \\ \Gamma^y_{yz} &= \frac{y^2z}{x^2+y^2+z^2+1} \\ \Gamma^y_{zz} &= -\frac{y^3+(x^2+1)y}{x^2+y^2+z^2+1} \\ \Gamma^z_{xx} &= -\frac{z^3+(y^2+1)z}{x^2+y^2+z^2+1} \\ \Gamma^z_{xy} &= \frac{xyz}{x^2+y^2+z^2+1} \\ \Gamma^z_{xz} &= \frac{xz^2}{x^2+y^2+z^2+1} \\ \Gamma^z_{yy} &= -\frac{z^3+(x^2+1)z}{x^2+y^2+z^2+1} \\ \Gamma^z_{yz} &= \frac{yz^2}{x^2+y^2+z^2+1} \\ \Gamma^z_{zz} &= -\frac{(x^2+y^2+1)z}{x^2+y^2+z^2+1} \end{aligned}$$

```
[257]: nabla.display(frame=Y.frame(), chart=Y, only_nonredundant=True)
```

```
[257]:
```

$$\begin{aligned} \Gamma^r_{rr} &= -\frac{r}{r^2+1} \\ \Gamma^r_{\theta\theta} &= -r^3 - r \\ \Gamma^r_{\phi\phi} &= -(r^3 + r) \sin(\theta)^2 \\ \Gamma^\theta_{r\theta} &= \frac{1}{r} \\ \Gamma^\theta_{\phi\phi} &= -\cos(\theta) \sin(\theta) \\ \Gamma^\phi_{r\phi} &= \frac{1}{r} \\ \Gamma^\phi_{\theta\phi} &= \frac{\cos(\theta)}{\sin(\theta)} \end{aligned}$$

Rimann テンソルは、今や：

The Riemann tensor is now

```
[258]: Riem = nabra.riemann()
print(Riem) ; Riem.display(Y.frame())
```

Tensor field Riem(g) of type (1,3) on the 3-dimensional differentiable manifold M

[258]:

$$\begin{aligned} \text{Riem}(g) = & -r^2 \frac{\partial}{\partial r} \otimes d\theta \otimes dr \otimes d\theta + r^2 \frac{\partial}{\partial r} \otimes d\theta \otimes d\theta \otimes dr - r^2 \sin(\theta)^2 \frac{\partial}{\partial r} \otimes d\phi \otimes dr \otimes d\phi + r^2 \sin(\theta)^2 \frac{\partial}{\partial r} \otimes \\ & d\phi \otimes d\phi \otimes dr + \left(\frac{1}{r^2+1}\right) \frac{\partial}{\partial \theta} \otimes dr \otimes dr \otimes d\theta + \left(-\frac{1}{r^2+1}\right) \frac{\partial}{\partial \theta} \otimes dr \otimes d\theta \otimes dr - r^2 \sin(\theta)^2 \frac{\partial}{\partial \theta} \otimes d\phi \otimes \\ & d\theta \otimes d\phi + r^2 \sin(\theta)^2 \frac{\partial}{\partial \theta} \otimes d\phi \otimes d\phi \otimes d\theta + \left(\frac{1}{r^2+1}\right) \frac{\partial}{\partial \phi} \otimes dr \otimes dr \otimes d\phi + \left(-\frac{1}{r^2+1}\right) \frac{\partial}{\partial \phi} \otimes dr \otimes \\ & d\phi \otimes dr + r^2 \frac{\partial}{\partial \phi} \otimes d\theta \otimes d\theta \otimes d\phi - r^2 \frac{\partial}{\partial \phi} \otimes d\theta \otimes d\phi \otimes d\theta \end{aligned}$$

リーマンテンソルは接続などを明示的に求めることなく、計量  $g$  から直接算出することもできます。

Note that it can be accessed directly via the metric, without any explicit mention of the connection:

```
[259]: g.riemann() is nabra.riemann()
```

[259]: True

```
[260]: g.riemann().display(Y.frame())
```

[260]:

$$\begin{aligned} \text{Riem}(g) = & -r^2 \frac{\partial}{\partial r} \otimes d\theta \otimes dr \otimes d\theta + r^2 \frac{\partial}{\partial r} \otimes d\theta \otimes d\theta \otimes dr - r^2 \sin(\theta)^2 \frac{\partial}{\partial r} \otimes d\phi \otimes dr \otimes d\phi + r^2 \sin(\theta)^2 \frac{\partial}{\partial r} \otimes \\ & d\phi \otimes d\phi \otimes dr + \left(\frac{1}{r^2+1}\right) \frac{\partial}{\partial \theta} \otimes dr \otimes dr \otimes d\theta + \left(-\frac{1}{r^2+1}\right) \frac{\partial}{\partial \theta} \otimes dr \otimes d\theta \otimes dr - r^2 \sin(\theta)^2 \frac{\partial}{\partial \theta} \otimes d\phi \otimes \\ & d\theta \otimes d\phi + r^2 \sin(\theta)^2 \frac{\partial}{\partial \theta} \otimes d\phi \otimes d\phi \otimes d\theta + \left(\frac{1}{r^2+1}\right) \frac{\partial}{\partial \phi} \otimes dr \otimes dr \otimes d\phi + \left(-\frac{1}{r^2+1}\right) \frac{\partial}{\partial \phi} \otimes dr \otimes \\ & d\phi \otimes dr + r^2 \frac{\partial}{\partial \phi} \otimes d\theta \otimes d\theta \otimes d\phi - r^2 \frac{\partial}{\partial \phi} \otimes d\theta \otimes d\phi \otimes d\theta \end{aligned}$$

**Ricci** テンソル はこのようになります。

The Ricci tensor is

```
[261]: Ric = g.ricci()
print(Ric)
Ric.display(Y.frame())
```

Field of symmetric bilinear forms Ric(g) on the 3-dimensional differentiable manifold M

[261]:

$$\text{Ric}(g) = \left(-\frac{2}{r^2+1}\right) dr \otimes dr - 2r^2 d\theta \otimes d\theta - 2r^2 \sin(\theta)^2 d\phi \otimes d\phi$$

**Weyl** テンソル は :

The Weyl tensor is:

Weyl テンソルについては Weinberg Page 145, 内山 page. 322, eq. (52.25) を参照

```
[262]: C = g.weyl()
print(C) ; C.display()
```

Tensor field  $C(g)$  of type  $(1,3)$  on the 3-dimensional differentiable manifold  $M$

```
[262]: C(g) = 0
```

と、多様体  $\mathcal{M}$  の次元が 3 なので、**Weyl** テンソルは恒等的に 0 になります。

最後に **Ricci** スカラ (曲率) を求めます。

The Weyl tensor vanishes identically because the dimension of  $\mathcal{M}$  is 3.

Finally, the Ricci scalar is

```
[263]: R = g.ricci_scalar()
print(R) ; R.display()
```

Scalar field  $r(g)$  on the 3-dimensional differentiable manifold  $M$

```
[263]: r(g): M      -> R
      (x,y,z)  -> -6
on U: (r,theta,phi) -> -6
```

このようにこの空間  $\mathbb{H}^3$  は負の定曲率をもった空間 (双曲空間) であることが、改めて確認できます。

We recover the fact that  $\mathbb{H}^3$  is a Riemannian manifold of constant negative curvature.

### 1.13 計量によって引き起こされるテンソル場の変形: Tensor transformations induced by a metric

もっとも重要な軽量  $g$  によって引き起こされるテンソル場の変形は、**musical isomorphism** あるいは **指標の引き上げおよび 指標の引き下げ** と呼ばれるものです。

The most important tensor transformation induced by the metric  $g$  is the so-called musical isomorphism, or index raising and index lowering:

```
[264]: print(t)
```

Tensor field  $T$  of type  $(1,2)$  on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

```
[265]: t.display()
```

```
[265]: T = (r^2 cos(phi)^2 sin(theta)^2 + 1) * partial_x dy otimes dx + r^3 cos(phi) cos(theta) sin(phi) sin(theta)^2 * partial_z dy otimes dx
```

```
[266]: t.display(X_U.frame(), X_U)
```

```
[266]: T = (x^2 + 1) * partial_x dy otimes dx + xyz * partial_z dy otimes dx
```

テンソル場  $T$  の最後の指標 (位置は 2) を計量  $g$  について、引き上げます。

Raising the last index (position 2) of  $T$  with  $g$ :

```
[267]: s = t.up(g, 2)
print(t,s)
```

Tensor field  $T$  of type (1,2) on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$  Tensor field of type (2,1) on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

上記の式で引き上げられた指標は反変な指数のうち最後のものになることに注意しましょう。つまり、メソッド `up` が返すテンソル  $s$  は

$$s^{ab}{}_c = g^{bi} T^a{}_{ic}$$

です。

Note that the raised index becomes the *last* one among the contravariant indices, i.e. the tensor  $s$  returned by the method `up` is

$$s^{ab}{}_c = g^{bi} T^a{}_{ic}$$

これについての詳細は [up\(\)](#) の解説をご覧ください。

See the [up\(\)](#) documentation for more details.

$T$  の全ての共変な指標 (この例では 位置 1 と 2) を引き上げます。

Raising all the covariant indices of  $T$  (i.e. those at the positions 1 and 2):

```
[268]: s = t.up(g)
print(s)
```

Tensor field of type (3,0) on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

```
[269]: s = t.down(g)
print(s)
```

Tensor field of type (0,3) on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

上記の式で引き下げられた指標は 共変な指標のうち 最初の指標になることに注意しましょう。 即ち、`down` メソッドが返すテンソル  $s$  は、

$$s_{abc} = g_{ai} T^i{}_{bc}$$

となります。

Note that the lowered index becomes the *first* one among the covariant indices, i.e. the tensor  $s$  returned by the method `down` is

$$s_{abc} = g_{ai} T^i{}_{bc}$$

これについての詳細は [down\(\)](#) の解説をご覧ください。

See the [down\(\)](#) documentation for more details.



## 1.14 Hodge 双対 (星印演算子) : Hodge duality

計量  $g$  についての体積 3-形式 (Levi-Civita テンソル ) は

The volume 3-form (Levi-Civita tensor) associated with the metric  $g$  is

```
[270]: epsilon = g.volume_form()
print(epsilon)
epsilon.display()
```

3-form  $\epsilon_g$  on the 3-dimensional differentiable manifold  $M$

```
[270]: 
$$\epsilon_g = \left( \frac{1}{\sqrt{x^2 + y^2 + z^2 + 1}} \right) dx \wedge dy \wedge dz$$

```

```
[271]: epsilon.display(Y.frame())
```

```
[271]: 
$$\epsilon_g = \left( \frac{r^2 \sin(\theta)}{\sqrt{r^2 + 1}} \right) dr \wedge d\theta \wedge d\phi$$

```

```
[272]: epsilon.display(Y.frame())
```

```
[272]: 
$$\epsilon_g = \left( \frac{r^2 \sin(\theta)}{\sqrt{r^2 + 1}} \right) dr \wedge d\theta \wedge d\phi$$

```

```
[273]: print(f) ; f.display()
```

Scalar field  $f$  on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

```
[273]: 
$$\begin{aligned} f: U &\longrightarrow \mathbb{R} \\ (x, y, z) &\longmapsto z^3 + y^2 + x \\ (r, \theta, \phi) &\longmapsto r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r \cos(\phi) \sin(\theta) \end{aligned}$$

```

```
[274]: sf = f.hodge_dual(g.restrict(U))
print(sf)
sf.display()
```

3-form  $\star f$  on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

```
[274]: 
$$\star f = \left( \frac{r^3 \cos(\theta)^3 + r^2 \sin(\phi)^2 \sin(\theta)^2 + r \cos(\phi) \sin(\theta)}{\sqrt{r^2 + 1}} \right) dx \wedge dy \wedge dz$$

```

古典的な式  $\star f = f \epsilon_g$ , より正確には  $\star f = f \epsilon_g|_U$ , を確かめてみましょう。  $f$  は  $U$  上でのみ定義されていることに注意します。

We check the classical formula  $\star f = f \epsilon_g$ , or, more precisely,  $\star f = f \epsilon_g|_U$  (for  $f$  is defined on  $U$  only):

```
[275]: sf == f * epsilon.restrict(U)
```

```
[275]: True
```

1-形式の Hodge 双対は、2-形式です。(3次元多様体の場合)；

The Hodge dual of a 1-form is a 2-form:

```
[276]: print(om) ; om.display()
```

1-form omega on the Open subset U of the 3-dimensional differentiable manifold M

```
[276]: ω = r2 sin(θ)2 dx + r cos(θ) dy + (r cos(φ) sin(θ) - r cos(θ)) dz
```

```
[277]: som = om.hodge_dual(g)
print(som) ;
som.display()
```

2-form \*omega on the Open subset U of the 3-dimensional differentiable manifold M

```
[277]:
```

$$\begin{aligned} *ω = & \left( \frac{r^4 \cos(\phi) \cos(\theta) \sin(\theta)^3 - r^3 \cos(\theta)^3 - r \cos(\theta) + (r^3(\cos(\phi) + \sin(\phi)) \cos(\theta)^2 + r \cos(\phi)) \sin(\theta)}{\sqrt{r^2 + 1}} \right) dx \wedge \\ & dy + \left( \frac{r^4 \cos(\phi) \sin(\phi) \sin(\theta)^4 - r^3 \cos(\theta)^2 \sin(\phi) \sin(\theta) + (\cos(\phi) \sin(\phi) + \sin(\phi)^2) r^3 \cos(\theta) \sin(\theta)^2 + r \cos(\theta)}{\sqrt{r^2 + 1}} \right) \\ & dz + \left( \frac{r^4 \cos(\phi)^2 \sin(\theta)^4 - r^3 \cos(\phi) \cos(\theta)^2 \sin(\theta) + ((\cos(\phi)^2 + \cos(\phi) \sin(\phi)) r^3 \cos(\theta) + r^2) \sin(\theta)^2}{\sqrt{r^2 + 1}} \right) dy \wedge \\ & dz \end{aligned}$$

(3次元多様体上では) 2-形式の hodge 双対は 1-形式になります。

The Hodge dual of a 2-form is a 1-form:

```
[278]: print(a)
```

2-form A on the Open subset U of the 3-dimensional differentiable manifold M

```
[279]: sa = a.hodge_dual(g)
print(sa)
sa.display()
```

1-form \*A on the Open subset U of the 3-dimensional differentiable manifold M

```
[279]:
```

$$\begin{aligned} *A = & \left( \frac{3r^5 \cos(\theta)^5 + 3r^3 \cos(\theta)^3 + (3r^6 \cos(\phi) \cos(\theta)^2 \sin(\phi) - 2r^5 \cos(\phi) \cos(\theta) \sin(\phi) - 2r^4 \cos(\phi) \sin(\phi)^3)}{\sqrt{r^2 + 1}} \right) dx \\ & \left( \frac{r^3 \cos(\theta)^3 + (3r^6 \cos(\phi)^2 \cos(\theta)^2 - 2(\cos(\phi)^2 - 1)r^5 \cos(\theta) + 2(\cos(\phi)^4 - \cos(\phi)^2)r^4) \sin(\theta)^4 - (r^3 \cos(\phi) \cos(\theta) \sin(\theta)^2 + r^2 \cos(\phi) \sin(\theta)^2)}{\sqrt{r^2 + 1}} \right) dy \\ & \left( \frac{2r^5 \sin(\phi) \sin(\theta)^5 + (3r^6 \cos(\theta)^3 \sin(\phi) + 2r^4 \cos(\phi)^2 \cos(\theta) \sin(\phi) + 2r^3 \sin(\phi)) \sin(\theta)^3 - (2r^4 \cos(\phi) \cos(\theta) \sin(\theta)^2 + r^3 \cos(\phi) \sin(\theta)^2)}{\sqrt{r^2 + 1}} \right) dz \end{aligned}$$

最後に 3-形式の Hodge 双対は 0-forms です。

Finally, the Hodge dual of a 3-form is a 0-form:

```
[280]: print(da)
       da.display()
```

3-form  $dA$  on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

```
[280]: dA = (-2 * (3 * r**3 * cos(theta)**2 * sin(phi) + r * sin(phi)) * sin(theta) - 1) * dx * dy * dz
```

```
[281]: sda = da.hodge_dual(g)
       print(sda)
       show(sda.display())
```

Scalar field  $*dA$  on the Open subset  $U$  of the 3-dimensional differentiable manifold  $M$

$$\begin{aligned} *dA: U &\longrightarrow \mathbb{R} \\ (x, y, z) &\longmapsto -(6yz^2 + 2y + 1)\sqrt{x^2 + y^2 + z^2 + 1} \\ (r, \theta, \phi) &\longmapsto -\sqrt{r^2 + 1} \left( 2 \left( 3r^3 \cos(\theta)^2 \sin(\phi) + r \sin(\phi) \right) \sin(\theta) + 1 \right) \end{aligned}$$

3次元のリーマン計量に対して、Hodge 双対は idempotent、つまり 2 回続けて作用させると元に戻ります。

In dimension 3 and for a Riemannian metric, the Hodge star is idempotent:

```
[282]: sf.hodge_dual(g) == f
```

```
[282]: True
```

```
[283]: som.hodge_dual(g) == om
```

```
[283]: True
```

```
[284]: sa.hodge_dual(g) == a
```

```
[284]: True
```

```
[285]: sda.hodge_dual(g.restrict(U)) == da
```

```
[285]: True
```

## 1.15 助けを求める : Getting help

オブジェクトが呼び出すことのできる関数 (メソッド) を入手するには、オブジェクトの名前を入力したのち、“.” と TAB を入力します。つまり、`sa.<tab>` です。

オブジェクトやメソッドについての情報を入手するために、疑問符?をお使いください。

To get the list of functions (methods) that can be called on a object, type the name of the object, followed by a dot and the TAB key, e.g.

sa.

To get information on an object or a method, use the question mark:

[286]: `nabla?`

**Signature:** `nabla(tensor)`  
**Type:** `LeviCivitaConnection`  
**String form:** Levi-Civita connection `nabla_g` associated with the Riemannian metric `g` on the 3-dimensional differentiable manifold `M`  
**File:** `/private/var/tmp/sage-10.0-current/local/var/lib/sage/venv-python3.11.1/lib/python3.11/site-packages/sage/manifolds/differentiable/levi_civita_connection.py`  
**Docstring:**  
Levi-Civita connection on a pseudo-Riemannian manifold.

Let  $M$  be a differentiable manifold of class  $C^\infty$  (smooth manifold) over  $\mathbb{R}$  endowed with a pseudo-Riemannian metric  $g$ . Let  $C^\infty(M)$  be the algebra of smooth functions  $M \rightarrow \mathbb{R}$  (cf. "DiffScalarFieldAlgebra") and let  $\mathfrak{X}(M)$  be the  $C^\infty(M)$ -module of vector fields on  $M$  (cf. "VectorFieldModule"). The \*Levi-Civita connection associated with\*  $g$  is the unique operator

$$\begin{array}{l} \nabla: \mathfrak{X}(M) \times \\ \mathfrak{X}(M) \times \mathfrak{X}(M) \rightarrow \mathfrak{X}(M) \\ (u, v) \mapsto \nabla_u v \end{array}$$

that

- \* is  $\mathbb{R}$ -bilinear, i.e. is bilinear when considering  $\mathfrak{X}(M)$  as a vector space over  $\mathbb{R}$
- \* is  $C^\infty(M)$ -linear w.r.t. the first argument:  $\forall f \in C^\infty(M), \nabla_{fu} v = f \nabla_u v$
- \* obeys Leibniz rule w.r.t. the second argument:  $\forall f \in C^\infty(M), \nabla_u (f v) = df(u) v + f \nabla_u v$
- \* is torsion-free
- \* is compatible with  $g$ :  $\forall (u, v, w) \in \mathfrak{X}(M)^3, \nabla_u (g(v, w)) = g(\nabla_u v, w) + g(v, \nabla_u w)$

The Levi-Civita connection  $\nabla$  gives birth to the \*covariant derivative operator\* acting on tensor fields, denoted by the same symbol:

```

\begin{array}{cccc} \nabla: & T^{\{k,l\}}(M) & \dashrightarrow & T^{\{k,l+1\}}(M) \\ & & & \\ & t & \dashrightarrow & \nabla t \\ \end{array}

```

where  $T^{\{k,l\}}(M)$  stands for the  $C^\infty(M)$ -module of tensor fields of type  $(k,l)$  on  $M$  (cf. "TensorFieldModule"), with the convention  $T^{\{(0,0)\}}(M) := C^\infty(M)$ . For a vector field  $v$ , the covariant derivative  $\nabla v$  is a type- $(1,1)$  tensor field such that

```

\forall u \in \mathfrak{X}(M), \quad \nabla_u v = \nabla v(\cdot, u)

```

More generally for any tensor field  $t \in T^{\{k,l\}}(M)$ , we have

```

\forall u \in \mathfrak{X}(M), \quad \nabla_u t = \nabla t(\dots, u)

```

Note:

The above convention means that, in terms of index notation, the "derivation index" in  $\nabla t$  is the *last* one:

$$\nabla_c t^{\{a_1 \dots a_k\}} \quad \nabla_c b_{1 \dots b_l} = (\nabla t)^{\{a_1 \dots a_k\}} \quad \nabla_c b_{1 \dots b_l}$$

INPUT:

- \* "metric" -- the metric  $g$  defining the Levi-Civita connection, as an instance of class "PseudoRiemannianMetric"
- \* "name" -- name given to the connection
- \* "latex\_name" -- (default: "None") LaTeX symbol to denote the connection
- \* "init\_coef" -- (default: "True") determines whether the Christoffel symbols are initialized (in the top charts on the domain, i.e. disregarding the subcharts)

EXAMPLES:

Levi-Civita connection associated with the Euclidean metric on  $\mathbb{R}^3$  expressed in spherical coordinates:

```

sage: forget() # for doctests only
sage: M = Manifold(3, 'R^3', start_index=1)
sage: c_spher.<r,th,ph> = M.chart(r'r:(0,+oo) th:(0,pi):\theta ph:(0,2*pi):
↪\phi')

```

```

sage: g = M.metric('g')
sage: g[1,1], g[2,2], g[3,3] = 1, r^2 , (r*sin(th))^2
sage: g.display()
g = dr ⊗ dr + r^2 dth ⊗ dth + r^2*sin(th)^2 dph ⊗ dph
sage: nab = g.connection(name='nabla', latex_name=r'\nabla') ; nab
Levi-Civita connection nabla associated with the Riemannian metric g on
the 3-dimensional differentiable manifold R^3

```

Let us check that the connection is compatible with the metric:

```

sage: Dg = nab(g) ; Dg
Tensor field nabla(g) of type (0,3) on the 3-dimensional
differentiable manifold R^3
sage: Dg == 0
True

```

and that it is torsionless:

```

sage: nab.torsion() == 0
True

```

As a check, let us enforce the computation of the torsion:

```

sage: sage.manifolds.differentiable.affine_connection.AffineConnection.
↪torsion(nab) == 0
True

```

The connection coefficients in the manifold's default frame are Christoffel symbols, since the default frame is a coordinate frame:

```

sage: M.default_frame()
Coordinate frame (R^3, (∂ / ∂ r, ∂ / ∂ th, ∂ / ∂ ph))
sage: nab.coef()
3-indices components w.r.t. Coordinate frame (R^3, (∂ / ∂ r, ∂ / ∂ th, ∂ / ∂
ph)),
with symmetry on the index positions (1, 2)

```

We note that the Christoffel symbols are symmetric with respect to their last two indices (positions (1,2)); their expression is:

```

sage: nab.coef()[:] # display as a array
[[[0, 0, 0], [0, -r, 0], [0, 0, -r*sin(th)^2]],
 [[0, 1/r, 0], [1/r, 0, 0], [0, 0, -cos(th)*sin(th)]],
 [[0, 0, 1/r], [0, 0, cos(th)/sin(th)], [1/r, cos(th)/sin(th), 0]]]
sage: nab.display() # display only the non-vanishing symbols
Gam^r_th,th = -r
Gam^r_ph,ph = -r*sin(th)^2

```

```

    Gam^th_r,th = 1/r
    Gam^th_th,r = 1/r
    Gam^th_ph,ph = -cos(th)*sin(th)
    Gam^ph_r,ph = 1/r
    Gam^ph_th,ph = cos(th)/sin(th)
    Gam^ph_ph,r = 1/r
    Gam^ph_ph,th = cos(th)/sin(th)
    sage: nab.display(only_nonredundant=True) # skip redundancy due to
↪symmetry
    Gam^r_th,th = -r
    Gam^r_ph,ph = -r*sin(th)^2
    Gam^th_r,th = 1/r
    Gam^th_ph,ph = -cos(th)*sin(th)
    Gam^ph_r,ph = 1/r
    Gam^ph_th,ph = cos(th)/sin(th)

```

The same display can be obtained via the function "christoffel\_symbols\_display()" acting on the metric:

```

    sage: g.christoffel_symbols_display(chart=c_spher)
    Gam^r_th,th = -r
    Gam^r_ph,ph = -r*sin(th)^2
    Gam^th_r,th = 1/r
    Gam^th_ph,ph = -cos(th)*sin(th)
    Gam^ph_r,ph = 1/r
    Gam^ph_th,ph = cos(th)/sin(th)

```

**Init docstring:** Construct a Levi-Civita connection.  
**Call docstring:** Action of the connection on a tensor field.

INPUT:

\* "tensor" -- a tensor field T, of type (k,\ell)

OUTPUT:

\* tensor field  $\nabla T$ .

[287]: `g.ricci_scalar?`

**Signature:** `g.ricci_scalar(name=None, latex_name=None)`

**Docstring:**

Return the metric's Ricci scalar.

The Ricci scalar is the scalar field  $r$  defined from the Ricci tensor  $\text{Ric}$  and the metric tensor  $g$  by

```
r = g^{ij} Ric_{ij}
```

INPUT:

- \* "name" -- (default: "None") name given to the Ricci scalar; if none, it is set to "r(g)", where "g" is the metric's name
- \* "latex\_name" -- (default: "None") LaTeX symbol to denote the Ricci scalar; if none, it is set to "r(g)", where "g" is the metric's name

OUTPUT:

- \* the Ricci scalar r, as an instance of "DiffScalarField"

EXAMPLES:

Ricci scalar of the standard metric on the 2-sphere:

```
sage: M = Manifold(2, 'S^2', start_index=1)
sage: U = M.open_subset('U') # the complement of a meridian (domain of ↵
↵spherical coordinates)
sage: c_spher.<th,ph> = U.chart(r'th:(0,pi):\theta ph:(0,2*pi):\phi')
sage: a = var('a') # the sphere radius
sage: g = U.metric('g')
sage: g[1,1], g[2,2] = a^2, a^2*sin(th)^2
sage: g.display() # standard metric on the 2-sphere of radius a:
g = a^2 dth ⊗ dth + a^2*sin(th)^2 dph ⊗ dph
sage: g.ricci_scalar()
Scalar field r(g) on the Open subset U of the 2-dimensional
differentiable manifold S^2
sage: g.ricci_scalar().display() # The Ricci scalar is constant:
r(g): U →
      (th, ph)    2/a^2
```

**Init docstring:** Initialize self. See help(type(self)) for accurate signature.  
**File:** /private/var/tmp/sage-10.0-current/local/var/lib/sage/venv-python3.11.1/lib/python3.11/site-packages/sage/manifolds/differentiable/metric.py  
**Type:** method

二つの疑問符 ?? は **Python** ソースコードに直接導かれます。(SageMath はオープンソースです、そうでしょう?)

Using a double question mark leads directly to the **Python source code** (SageMath is **open source**, isn't it?):

```
[288]: g.ricci_scalar??
```

**Signature:** g.ricci\_scalar(name=None, latex\_name=None)



**Docstring:**

Return the metric's Ricci scalar.

The Ricci scalar is the scalar field  $r$  defined from the Ricci tensor  $\text{Ric}$  and the metric tensor  $g$  by

$$r = g^{\{ij\}} \text{Ric}_{\{ij\}}$$

INPUT:

- \* "name" -- (default: "None") name given to the Ricci scalar; if none, it is set to "r(g)", where "g" is the metric's name
- \* "latex\_name" -- (default: "None") LaTeX symbol to denote the Ricci scalar; if none, it is set to "r(g)", where "g" is the metric's name

OUTPUT:

- \* the Ricci scalar  $r$ , as an instance of "DiffScalarField"

EXAMPLES:

Ricci scalar of the standard metric on the 2-sphere:

```
sage: M = Manifold(2, 'S^2', start_index=1)
sage: U = M.open_subset('U') # the complement of a meridian (domain of
←spherical coordinates)
sage: c_spher.<th,ph> = U.chart(r'th:(0,pi):\theta ph:(0,2*pi):\phi')
sage: a = var('a') # the sphere radius
sage: g = U.metric('g')
sage: g[1,1], g[2,2] = a^2, a^2*sin(th)^2
sage: g.display() # standard metric on the 2-sphere of radius a:
g = a^2 dth ⊗ dth + a^2*sin(th)^2 dph ⊗ dph
sage: g.ricci_scalar()
Scalar field r(g) on the Open subset U of the 2-dimensional
differentiable manifold S^2
sage: g.ricci_scalar().display() # The Ricci scalar is constant:
r(g): U →
      (th, ph)    2/a^2
```

**Source:**

```
def ricci_scalar(self, name=None, latex_name=None):
    r"""
    Return the metric's Ricci scalar.
```

```
    The Ricci scalar is the scalar field `r` defined from the Ricci tensor
    `Ric` and the metric tensor `g` by
```

```

.. MATH::

    r = g^{ij} Ric_{ij}

INPUT:

- ``name`` -- (default: ``None``) name given to the Ricci scalar;
  if none, it is set to "r(g)", where "g" is the metric's name
- ``latex_name`` -- (default: ``None``) LaTeX symbol to denote the
  Ricci scalar; if none, it is set to "\mathrm{r}(g)", where "g"
  is the metric's name

OUTPUT:

- the Ricci scalar `r`, as an instance of
  :class:`~sage.manifolds.differentiable.scalarfield.DiffScalarField`

EXAMPLES:

Ricci scalar of the standard metric on the 2-sphere::

    sage: M = Manifold(2, 'S^2', start_index=1)
    sage: U = M.open_subset('U') # the complement of a meridian (domain
↳ of spherical coordinates)
    sage: c_spher.<th,ph> = U.chart(r'th:(0,pi):\theta ph:(0,2*pi):\phi')
    sage: a = var('a') # the sphere radius
    sage: g = U.metric('g')
    sage: g[1,1], g[2,2] = a^2, a^2*sin(th)^2
    sage: g.display() # standard metric on the 2-sphere of radius a:
    g = a^2 dth ⊗ dth + a^2*sin(th)^2 dph ⊗ dph
    sage: g.ricci_scalar()
    Scalar field r(g) on the Open subset U of the 2-dimensional
    differentiable manifold S^2
    sage: g.ricci_scalar().display() # The Ricci scalar is constant:
    r(g): U →
        (th, ph)    2/a^2

"""
if self._ricci_scalar is None:
    manif = self._ambient_domain
    ric = self.ricci()
    ig = self.inverse()
    frame = ig.common_basis(ric)
    cric = ric._components[frame]
    cig = ig._components[frame]
    rsum1 = 0

```

```

for i in manif.irange():
    rsum1 += cig[[i,i]] * cric[[i,i]]
rsum2 = 0
for i in manif.irange():
    for j in manif.irange(start=i+1):
        rsum2 += cig[[i,j]] * cric[[i,j]]
self._ricci_scalar = rsum1 + 2*rsum2
if name is None:
    self._ricci_scalar._name = "r(" + self._name + ")"
else:
    self._ricci_scalar._name = name
if latex_name is None:
    self._ricci_scalar._latex_name = r"\mathrm{r}\left(" + \
        self._latex_name + r"\right)"
else:
    self._ricci_scalar._latex_name = latex_name
return self._ricci_scalar

```

File: /private/var/tmp/sage-10.0-current/local/var/lib/sage/venv-python3.11.  
↳1/lib/python3.11/site-packages/sage/manifolds/differentiable/metric.py  
Type: method

## 1.16 さらに先へ : Going further

ページ:[SageManifolds の例](#)をご覧ください。とくに、平行化不能な多様体（スカラー場は少なくとも二つの座標系で定義される必要があり、モジュール  $\mathfrak{X}(\mathcal{M})$  はもはやフリーではなく、テンソル場は少なくとも二つのフレームで定義される必要があります）での使い方については、[2次元球面](#) が役に立つでしょう。

Christian Bär 氏による[チュートリアル](#) [ビデオ](#)もご覧ください。

Have a look at the [examples on SageManifolds page](#), especially the [2-dimensional sphere](#) for usage on a non-parallelizable manifold (each scalar field has to be defined in at least two coordinate charts, the  $C^\infty(\mathcal{M})$ -module  $\mathfrak{X}(\mathcal{M})$  is no longer free and each tensor field has to be defined in at least two vector frames).

You may also take a look at the [tutorial videos](#) by Christian Bär.

[ ]: